
Avant-propos

Voici les sujets de TP d’informatique qui ont été proposés à l’E.D.E.N. lors de
l’année scolaire 2014-2015 en ECS1. Il s’agit d’un travail ex nihilo, un « premier
jet » certainement perfectible. J’ai corrigé certaines fautes, mais il doit en rester.
Les choix pédagogiques sont certainement discutables, d’ailleurs, je le referai(s)
différemment aujourd’hui. Je n’ai toutefois pas le temps de reprendre avec ce que
les élèves m’ont appris de ce qui était clair pour eux ou non. J’en profite pour les
remercier pour leurs nombreux retours.

Je rends texte disponible en espérant qu’il puisse être de quelque utilité, soit pour
l’apprentissage de Scilab , soit pour l’apprentissage de CONTEXT. C’est aussi en
hommage à tous celles et ceux qui partagent leurs connaissances sur la toile que
je partage cette contribution. Le contenu de ce document leur doit beaucoup et, si
je n’ai pas toujours noté mes sources au fur et à mesure, je tiens à remercier tous
ces mécènes.

Le texte a été mis en forme en TEX, plus exactement avec CONTEXT, surcouchemoins
connue en France que LATEXmais qui offre plus de possibilités de personnalisation
des documents. Si vous êtes intéressés par les sources, je vous les communiquerai
sur simple demande en espérant un retour qui puisse m’aider à améliorer le conte-
nu des TP et/ou le code CONTEXT

Je vous souhaite une bonne lecture.

Jean-Marc Berrou

TP1 | Découverte de Scilab

1.1 But du cours et découverte du logiciel 4
1.2 Premières opérations 5
1.3 Logique et booléens 7
1.4 Chaînes de caractères 9
1.5 La vraie nature de Scilab 10
1.6 Un peu de dessin 11

1.1 But du cours et découverte du logiciel

Le but de ces TP d’informatique est de se familia-
riser avec le logiciel Scilab , sa syntaxe, et la créa-
tion d’algorithmes simples dans ce langage en
lien avec le programme de mathématiques. Lors
des écrits de mathématiques, une ou plusieurs
questions d’algorithmique seront posées et il fau-
dra y répondre sur papier en langage Scilab . Ces
questions ne sont en général pas les plus difficiles
pour qui y est préparé et la meilleure prépara-
tion est de pratiquer régulièrement.

Vous pouvez installer le logiciel Scilab sur votre
ordinateur personnel car c’est un logiciel libre,
distribué gratuitement sur le site. Vous pouvez
l’utiliser comme une ’super calculatrice’ pour vé-
rifier vos calculs, faire des simulations….

Il est temps maintenant de lancer Scilab (et
attendre les instructions du professeur)

Figure 1.1 : La fenêtre principale de Scilab . Celle-ci est divisée en plusieurs zones

Celle de gauche est un navigateur qui permet
de retrouver vos fichiers (cf prochains TP). Au
centre, la partie où l’on tape les commandes et où
s’affichent les résultats. C’est celle que l’on utili-
sera le plus. En haut à droite, un récapitulatif de

toutes les variables utilisées, avec leur valeur ac-
tuelle. En dessous, l’historique des commandes. Si
vous avez besoin de retrouver une commande dé-
jà tapée, pas besoin de le refaire, elle a été conser-
vée en mémoire et est disponible dans ce cadre.

1.2 Premières opérations
On peut utiliser Scilab comme une calculatrice, pour faire des opérations simples.
C’est un bon moyen de se familiariser avec les principales commandes et peut
vous permettre plus tard de vérifier vos exercices (intégrales, matrices…) tout en
révisant l’informatique.

NB : En cas d’erreur,
vous pouvez retrouver
la dernière frappe (pour
la modifier) avec la
flèche ’haut’ ou dans la
zone d’historique en bas
à droite.

Exercice 1 Scilab comme ’calculatrice’

1. Effectuer les calculs suivants
(certaines commandes ne fonctionnent pas, c’est délibéré…)

--> 1+1
--> 1+8*9
--> 2^3
--> 45/7+2
--> 1/3*3
--> sin0
--> sin(0)
--> sin(%pi/2)
--> cos(%pi/12)
--> tan(%pi/4)
--> ans+1
--> %e
--> ln(%e)
--> (%i+1)^2
--> %e^(%i*%pi/4)
--> rand()

2. Taper help log. Sur la page d’aide, vous trouverez les commandes qui per-
mettent de calculer ln(3) et e5

3. La touche F1 donne accès à l’aide générale. Prenez un moment pour voir la
richesse du logiciel en attendant de passer à la suite.

À retenir

⋆ La commande help permet d’obtenir de l’aide sur n’importe commande.
Vous pouvez aussi utiliser la touche F1 .

⋆ Les principales constantes mathématiques s’obtiennent avec %
par exemple, %pi, %e, %i

⋆ la variable ans (comme answer) contient le résultat du dernier calcul.

Quelques fonctions usuelles de Scilab :

⋆ Racine carrée : sqrt()
⋆ Module : abs()
⋆ Partie entière : floor()
⋆ Logarithme népérien : log()
⋆ Logarithme en base 10 : log10()
⋆ Exponentielle : exp()
⋆ Cosinus : cos()
⋆ Sinus : sin()
⋆ Tangente : tan()
⋆ Arccos : acos()
⋆ Arcsin : asin()
⋆ Arctan : atan()

La commande rand() renvoie un nombre aléatoire entre 0 et 1

Exercice 2 utiliser la commande rand pour générer un nombre entier aléatoire
compris entre 10 et 20

Exercice 3 Affectations, utilisation de variables

Tapez les instructions suivantes dans la console en anticipant mentalement l’affi-
chage. Regardez aussi de temps en temps les fenêtres de droites pour voir ce qui
s’y passe.

--> a=2
--> a
--> a=5 ;
--> a
--> a=a+3
--> a=a+3;
--> a
--> b=a
--> c=a-3*b
--> a, b, c
--> a=3; b=2; c=a*b;
--> c
--> a+a+a+..
--> 1
--> 1+..
--> 2
--> who

Vous pourrez ensuite cliquer dans la fenêtre en bas à droite sur a=3; b=2; c=a*b;
plusieurs fois. Cela permet de rappeler la commande sans tout retaper. On peut
également utiliser la flèche `haut’ du clavier.

On voit de plus en plus
souvent dans les livres
récents de mathéma-
tiques l’utilisation de :=
pour définir un nouvel
objet de mathématique,
alors que= est utilisé
pour les identités et les
équations.

Attention

⋆ Le symbole = dans Scilab n’a pas le sens usuel du=mathématique. Il repré-
sente en fait la flèche← d’affectation. Scilab effectuera toujours le calcul
à droite du =, avec les valeurs en cours avant de rentrer le résultat dans la
variable à gauche du =

⋆ Noter la différence entre , et ; qui peut être importante si l’on fait tourner
un programme qui effectue des milliers de calculs…

⋆ On peut taper plusieurs calculs sur une même ligne en les séparant par ,
ou ;

⋆ Si on souhaite répartir un calcul sur deux lignes,
on peut le « découper » avec ..

Exercice 4 Entrez votre année de naissance dans la variable an et écrivez un
calcul avec cette variable.
Complément : si vous avez fini, sauriez-sous trouver une commande qui soit encore
valable dans quelques années ? (vous pourrez taper help date pour des idées
supplémentaires)

Conseil

Dans les programmes, préférez des noms de variables explicites, comme an,
age, moyenne, plutôt que des lettres n, N, x.
Bien que plus long à taper, cela facilite la lecture.

Exercice 5 Supposons que a et b sont deux variables déjà définies dans Scilab .
Donner une suite d’instructions qui échange les valeurs des variables a et b, puis
fait afficher les nouvelles valeurs de a et b.
Tester votre programme (prendre par exemple a=5 et b=3).

1.3 Logique et booléens

Exercice 6 : Logique et Booléens
Entrez les commandes suivantes :

--> a=2
--> a==2
--> a==5
--> a<5
--> a>5
--> c=a>5
--> c
--> b=%T
--> b|c
--> b&c

--> ~(2==3)

⋆ Scilab permet d’effectuer des tests logiques, dont la réponse est vrai (, true)
ou faux (, false). Cela sera utile dans les algorithmes pour introduire des
conditions avec if ou while…

⋆ On peut combiner les conditions avec les connecteurs et (noté &) et ou (noté
| obtenu avec AltGr+6 ) ainsi que la négation ~ (AltGr+2 )

⋆ Les valeurs vrai (, true) ou faux ( peuvent être affectées à des variables avec
et

Exercice 7 Prévoir les réponses de Scilab aux commandes suivantes :

--> (1>2)|(8<9)|(2<>5)
--> (abs(3+4*%i)>4)&((2==2)|(3==4))
--> ~((%T)|(%F))
--> ~((%T)&(%F))
--> ~((%T)|((%F)&(abs(2-3*%i)>log(%e 4))))

Le == est un test. Il
renvoie vrai ou faux. Il
correspond par exemple
au= des équations en
mathématiques.

Les principaux tests avec Scilab

⋆ égalité == ou différence <>

≠ se code en losange
car >< est plutôt un
signe de colère en émo-
ticones…

⋆ inégalité stricte < et >

⋆ inégalité large <= et >=

NB : Nous avons donc bien deux notations pour le = habituellement utilisé en
mathématiques :

⋆ le = pour l’affectation et
⋆ le == pour tester l’égalité.



TP1 | Découverte de Scilab

9 Jean-Marc BERROU - 2014-2015

1.4 Chaînes de caractères

Exercice 8 Observer l’effet des commandes suivantes :

--> a='bonjour'
--> a+a
--> a=a+' '
--> a
--> a=a+a
--> 'bonjour', 'bonjour'
--> A='Affecter à une variable une chaîne de caractères, ..
--> c''est simple ..
--> comme ''bonjour'' '
--> a+A
--> length(a+A)
--> part(a,3)
--> part(a,3:7)
--> clear
--> a

À retenir

⋆ on définit une chaîne de caractères avec des guillemets ’(3) ou des apos-
trophes ’’(4). Ces caractères ont donc un statut spécial et pour les afficher
il faut les doubler.

⋆ + ici réalise une concaténation. Les chaînes de caractères sont accolées (conca-
ténées)

⋆ On peut extraire une partie d’une chaîne de caractère avec la commande
part

⋆ De même que pour les vecteurs, la commande length() renvoie la lon-
gueur de la chaîne

1.5 La vraie nature de Scilab

Scilab est en fait bien plus qu’une
calculatrice. Il est utilisé en ingénie-
rie ou en finance pour des simula-
tions très avancées. C’est une sorte
d’expert en gros calculs.

Pour cela, les outils de base dans Sci-
lab ne sont pas les nombres mais
les vecteurs (ou plutôt matrices). Sci-
lab peut effectuer une même opéra-
tion sur toutes les composantes d’un
vecteur.

Pour définir un vecteur composante
par composante (cf les exemples ci
après) ou avec la commande suivante
très pratique :

nombre1:pas:nombre2

pour laquelle quelques exemples valent
mieux qu’un long discours.

Exercice 9 Vecteurs
Entrez les commandes suivantes

--> 1:1:5
--> 1:5
--> 1:2:5
--> 1:3:5
--> a=1:0.1:10
--> a^2
--> exp(a)
--> sin(a)
--> b=1:5
--> a+b
--> b+b
--> a=3:7
--> a+b
--> a+exp(b)
--> help sum
--> sum(a)
--> length(a)
--> a'
--> a=[1,2,3]
--> a=[1;2;3]
--> a=a'

Premiers pas avec les vecteurs

⋆ nombre1:pas:nombre2 crée le vecteur dont les coordonnées sont tous les
nombres de nombre1 inclus à nombre2 en allant de pas en pas

⋆ les vecteurs peuvent être très grands et permettre d’effectuer plusieurs opé-
rations en une ligne de commande

⋆ on peut effectuer la somme de toutes les composantes d’un vecteur avec
length(), calculer la somme de toutes les composantes avec sum(), trans-
poser un vecteur ligne en vecteur colonne avec le '

Exercice 10

1. Calculez avec Scilab la somme des 100 premiers carrés de nombres entiers

2. Calculez avec Scilab la somme des 1000 premiers cubes de nombres entiers

3. Calculez
100
∑
k=1

sin(k)

Le deuxième calcul nous rappelle que Scilab calcule des valeurs approchées,
c’est l’une des raisons de sa rapidité d’exécution. Un logiciel de calcul formel
comme Xcas ou Sage pourrait donner une valeur exacte.

1.6 Un peu de dessin
Si jamais rien ne
s’affiche, essayer :
usecanvas(%T)
dans la console puis
réessayer

Scilab dispose également de commandes graphiques qui rendent mieux grâce aux
nombres que des listes interminables.

Vous aurez certainement besoin de la commande
clf() (comme clear fenêtre ^^) qui efface la fenêtre graphique

Exercice 11

1. créez un vecteur x=-10:10

2. en tapant [x,x^2], vous obtenez un tableau de valeurs de la fonction carré

3. tapez maintenant plot(x,x^2) et vous obtenez la courbe de la fonction carré.
Vous auriez pu taper plot(x^2)

4. comparez
x=-10:10;plot(sin(x)) et
x=-10:0.1:10;plot(sin(x))

5. en cas de problème d’affichage, pensez à fermer la fenêtre ou créer un nouveau
graphique.

Exercice 12
Trouvez une commande qui affiche un cercle de centre 0 et de rayon 1.

Si vous avez tout terminé, vous pouvez taper help plot pour voir les différentes
options de tracé.

TP2 | Structures conditionnelles

Pour les concours, il vous sera demandé de comprendre le fonctionnement d’algo-
rithmes, d’être capables d’en écrire ou d’en modifier. Ceci implique deux compé-
tences distinctes :

⋆ comprendre les ressorts généraux de l’algorithmique d’une part
⋆ connaître la syntaxe de Scilab d’autre part.

Pour ce deuxième TP, nous allons voir comment écrire des algorithmes avec
Scilab

2.1 Quelques rappels d’algorithmique 12
2.2 Bien communiquer, c’est important : entrées et sorties 14
2.3 Enregistrer son travail : Scinotes 14
2.4 Structures conditionnelles if ... then ... else 15
2.5 Quelques corrigés : 18

2.1 Quelques rappels d’algorithmique
Un algorithme est une suite finie d’instructions dont l’exécution dans l’ordre
aboutit à la réalisation d’une tâche prédéfinie.

Un algorithme comprend en général :

⋆ des entrées
⋆ un traitement (les différentes instructions)
⋆ des sorties

2.1 au pluriel, ce n’est
pas l’entrée du menu ici

Un exemple bien connu d’algorithme est la recette de cuisine. En entrées1 nous
avons un certain nombre d’ingrédients. Le traitement consiste en les différentes
étapes d’élaboration.

La sortie est le plat finalisé. Si cet exemple paraît naïf, cette logique algorithmique
est poussée à l’extrême dans les usines agroalimentaires… et de manière générale
la pensée algorithmique est l’un des ressorts du monde industriel rationnalisé.

Pour être un peu plus précis sur ce qui fait qu’un ensemble de commandes est,
ou non, un algorithme, voici les cinq critères que l’éminent informaticien Donald

2.2 professeur émérite
de l’université de Stan-
ford né en 1938, connu
comme l’auteur de The
art of computer pro-
gramming, un des rares
ouvrages d’informa-
tique des années 70 qui
soit encore une réfé-
rence aujourd’hui. Il
est également l’inven-
teur des langages TEX
et MetaPost qui servent
encore aujourd’hui à la
conception des livres de
mathématiques du com-
merce et des documents
plus modestes comme
ce polycopié

Knuth2 a énoncés :

⋆ lafinitude : « Un algorithme doit toujours se terminer après un nombre
fini d’étapes. »

⋆ définition précise : « Chaque étape d’un algorithme doit être définie
précisément, les actions à transposer doivent être spécifiées rigoureu-
sement et sans ambiguïté pour chaque cas. »

⋆ entrées : « … des quantités qui lui sont données avant qu’un algo-
rithme ne commence. Ces entrées sont prises dans un ensemble d’objets
spécifié. »

⋆ sorties : « … des quantités ayant une relation spécifiées avec les en-
trées. »

⋆ rendement : « … toutes les opérations que l’algorithme doit accom-
plir doivent être suffisamment basiques pour pouvoir être en principe
réalisées dans une durée finie par un homme utilisant un papier et un
crayon. »

Les algorithmes sont souvent représentés par un organigramme comme les exemples
ci-dessous :

Tableau 2.1 :

Exercice 13 À quoi servent les deux algorithmes précédents ? En quoi le second
(algorithme de descente) n’est-il pas un algorithme au sens de Knuth ?

Certains langages,
comme le C ou le Pas-
cal demandent en plus
de ’préparer le maté-
riel’, c.a.d. d’annoncer
quels seront les espaces
de mémoires à allouer,
et pour quel type de
données. Scilab , est un
langage de haut niveau,
le plus bas niveau étant
le langage machine,
le plus haut niveau la
langue naturelle par-
lée et il s’adapte à ce
qu’on lui propose. Il
gère la mémoire au fil
des instructions.

2.2 Bien communiquer, c’est important :
entrées et sorties

La communication avec une machine, ce sont les entrées et sorties de l’algorithme.
Il faut donc bien retenir les commandes qui permettent à un programme de de-
mander à l’utilisateur d’entrer une valeur, une chaîne… et les commandes qui
permettent au programme d’afficher quelque chose (phrase, valeur, graphique ou
combinaison de tout cela).
La commande qui permet au programme de demander quelque chose à l’utilisateur
est input.

Par exemple, tapez en console :

n=input('entrez un nombre entier : ')

Qu’elle existât antérieurement ou non, la variable n aura désormais la valeur que
vous lui aurez attribuée en répondant à cet input. Notez également que si vous ne
souhaitez pas confirmation que la valeur a été enregistrée, vous pouvez terminer
la commande par ;

Figure 2.1:
La syntaxe, retenir tu
dois

Scilab peut afficher du texte, des variables avec disp.
Vous pouvez essayer :

--> disp(2*n,'le double du nombre est : ')
--> disp('le double du nombre est : ', 2*n)
--> disp([1,2],3)

Vous aurez remarqué que les variables sont affichées
de la dernière à la première. C’est un point qui peut
surprendre, et donc à bien retenir.

2.3 Enregistrer son travail : Scinotes
Dans le menu Applications de Scilab , vous trouverez Scinotes . Il s’agit d’un
éditeur de programme qui vous permettra de sauvegarder vos algorithmes et de les
retrouver au besoin. Vous pouvez par exemple taper les commandes précédentes
dans Scinotes pour en faire un programme :

n=input('entrez un nombre : ');
disp(2*n,'le double du nombre est :'), // attention à l'ordre inversé

Si vous appuyez sur ▷, le programme se lancera mais il vous sera demandé de le
sauvegarder avant exécution.

Dans votre espace personnel, créez un dossier TP_Scilab et placez vous dedans.
Choisissez ensuite comme nom de fichier TP1-exemple, le logiciel lui accolera
automatiquement l’extension .sce caractéristique des fichiers Scilab .

Quand on exécute le programme, la console Scilab affiche un message du genre
-->exec('/TP_Scilab/TP1-exemple.sce', -1) qui confirme que le programme
a bien été lancé.

Viennent ensuite le résultat des commandes effectuées au fil du programme.

Remarques

⋆ Retenez l’utilisation de // pour écrire des commentaires qui sont affi-
chés en italique vert par Scinotes

⋆ Un bon code est un code commenté. Ce qui vous paraît évident au-
jourd’hui ne le sera peut-être pas dans un mois !

⋆ On peut remarquer qu’à chaque modification, le script est enregistré
avant d’être exécuté, mais le nom n’est demandé qu’au premier lance-
ment. N’oubliez pas de changer de fichier quand vous créez un
nouveau programme ! En cas d’oubli, vous « écrasereriez » le fichier
de l’exercice n avec avec le script de l’exercice n + 1.
Ceci a deux fâcheuses implications : vous perdez le script n et le script
n + 1 est enregistré sous un faux nom ! Soyez vigilants.

⋆ Pensez à organiser vos scripts. On peut les organiser par nom ou
par dossier. Ici, je propose une organisation par nom. Chaque fichier
s’appellera TP2-ex3.sce par exemple ou TP2-exemple1.sce pour
le 3̀me exercice du TP n°2.

⋆ À l’instar des noms de variables, des noms de fichiers bien pensés
peuvent représenter un gain de temps pour une utilisation ultérieure.

⋆ Vous pouvez désormais enregistrer tous vos programmes.

Exercice 14
Écrire un programme qui demande un nombre a et renvoie la valeur de (a +
ln(1 + |a|))2
Certaines entrées donnent-elles un message d’erreur ?

2.4 Structures conditionnelles if ... then ...
else

Un algorithme ne consiste pas seulement en une suite d’instructions mises bout à
bout. Le comportement peut varier selon des cas prédéfinis à l’avance et effectuer
telle ou telle opération selon que telle ou telle condition est remplie. La syntaxe
Scilab correspondant est la suivante :

if (condition) then
Instruction 1
Instruction 2
...

elseif
Instruction 1
Instruction 2
...

else
Instruction 1
Instruction 2
...

end

Une condition ici est un booléen, c.a.d
une expression du type :

⋆ a==1 pour a = 1
⋆ a<1 pour a < 1
⋆ a<>1 pour a ≠ 1

⋆ (a>1)&(b<=2) a > 1 et b ≤ 2
⋆ (a==1)|(b>=2 pour a = 1 ou
b ≥ 2

Les else et elseif sont optionnels. Dans ce cas, il ne se passe rien si la condition
n’est pas remplie.
elseif permet d’ajouter un test si le premier n’est pas rempli.

Exercice 15 : Voici un programme

birth=input("Année de naissance (ex:1996): ");
year=getdate(); // getdate() renvoie un tableau avec
l’année
age=year(1)-birth;
if age >= 18 then

disp(" ans, vous êtes majeur !", age, "Vous avez ")
else

disp(" ans, vous êtes mineur !", age, "Vous avez ")
end

1. Lisez attentivement cet exemple pour en comprendre le fonctionnement, que
vous décrirez en une phrase.

2. Copier ce programme dans Scinotes , enregistrer sous age.sce. Exécuter le
programme plusieurs fois, en testant des entrées différentes.

3. Tester le programme avec une entrée supérieure à 2015 et le modifier le texte
du programme pour que ce genre de cas renvoie une phrase d’alerte ad hoc

Remarques

⋆ Remarquez l’indentation. Scinotes est conçu pour la proposer automati-
quement. Lors de vos écrits, n’oubliez pas d’indenter vos commandes
entre if ... then ... else

⋆ Le else n’est pas obligatoire si, par exemple, on souhaite exécuter des ins-
tructions uniquement si le test est vrai et ne rien faire sinon.

⋆ Il est tout à fait possible d’imbriquer plusieurs alternatives les unes à l’in-
térieur des autres.

Exercice 16 Racines carrées

Écrire un programme qui demande un nombre et renvoie sa racine carrée ou
message d’erreur ad hoc si l’entrée est un nombre négatif.

Exercice 17 Fonctions affines

Sauriez-vous écrire un algorithme qui prenne en entrée deux nombres a et b et

renvoie, en sortie, la résolution de l’équation ax + b = 0 ? L’algorithme devra
proposer une sortie adéquate y compris si a = 0.

Exercice 18 Second degré

Tout un chacun en ECS sait résoudre une équation du type ax2 + bx + c = 0.
C’est toujours la même chose n’est-ce pas ? Derrière chaque « c’est toujours la
même chose », se cache toujours un algorithme.
Sauriez-vous écrire un programme en Scilab , qui donnerait pour tous les cas (y
compris le cas a = 0) une phrase présentant les solutions éventuelles.

Exercice 19 Si vous êtes en avance…

Écrire un programme shifumi.sce … qui porte bien son nom

shi-fu-mi est le nom japonais du jeu « pierre, feuille, ciseaux ». Le programme devra donc afficher de
manière aléatoire l’un des trois mots « pierre », « feuille », ou « ciseaux ». Et si vous êtes vraiment
en avance, le programme pourrait afficher une invite vous demandant d’entrer 1, 2, ou 3 pour
« pierre », « feuille », ou « ciseaux » puis tire au hasard son propre tirage avant de décider du
vainqueur.
Si vous êtes toujours en avance, posez-vous la question : Ne rêveriez-vous pas secrètement de
travailler dans l’informatique ?

2.5 Quelques corrigés :

// TP2-ex3 ///////////////////////////////////////////
// majeur / mineur
birth=input("Année de naissance (ex:1996): ");
year=getdate();
// getdate() renvoie un tableau avec l’année
age=year(1)-birth;
if age<0 then // exclut le cas où l'age est négatif

disp(year(1),"vous ne pouvez pas être né après ")
else

if age >= 18 then
disp(" ans, vous êtes majeur !", age, "Vous avez ")

else
disp(" ans, vous êtes mineur !", age, "Vous avez ")

end
end

// TP2-ex4 //////////////////////////////////////////////
nombre=input("Donner le nombre dont vous voulez la racine
carrée : ");
if isreal(nombre)&(real(nombre)>=0)

disp(sqrt(nombre))
else

disp("le nombre rentré n''a pas de racine carrée")
end

// TP2-ex5 //////////////////////////////////////////////
disp("Veuillez entrer a puis b pour la résolution de l''équation
ax+b=0")
a=input("a=?")
b=input("b=?")
if a==0 then

if b==0 then
disp("tous les réels sont solution.")
disp("il y a une infinité de solutions")

else
disp("l''équation n''a pas de solution")

end
else

disp((-b/a),"l''équation admet pour unique solution :
")
end

Ici, pour tester le programme, il faut au moins que les trois cas a = b = 0, a = 0
et b = 1, a = 1 et b = 0 aient été envisagés.

// TP2-ex6 //////////////////////////////////////////////
disp("Veuillez entrer a, b puis c pour la résolution de l''équation ax^2+bx+c=0")
a=input("a=?")
b=input("b=?")
c=input("c=?")
if a==0 then

if b==0 then
if c==0 then

disp("tous les réels sont solution.")
disp("il y a une infinité de solutions")

else
disp("l''équation n''a pas de solution")

end
else

disp((-c/b),"l''équation admet pour unique solution : ")
end

else
delta=b^2-4*a*c
if delta >0 then

disp("l''équation admet deux solutions distinctes :")
disp((-b+sqrt(delta))/(2*a))
disp((-b-sqrt(delta))/(2*a),"et")

else
if delta==0 then

disp(-b/(2*a),"l''unique solution est :")
else

disp("l''équation n''admet pas de solutions réelles.")
end

end
end

Ici encore, tous les cas particuliers doivent avoir été envisagés, toutes les combi-
naisons de 0 possibles.

// TP2-ex7 //////////////////////////////////////////////
// Un programme de "shi-fu-mi"
x=floor(3*rand())
if x==0 then

disp('pierre')
else

if x==1 then
disp('feuille')

else
disp('ciseaux')

end
end

TP3 | Boucles for et while

L’objet de ce TP sera l’utilisation des boucles, ou structures itératives, dans vos
algorithmes. Au programme de ce TP :

3.1 Boucle for ... to ... end 20
3.2 Boucle while ... condition 25
3.3 Exercices 26
3.4 Quelques corrigés : 28

⋆ la boucle for est indiquée quand on sait à l’avance le nombre d’itérations
nécessaire.

⋆ la boucle while convient aux cas où on cherche à réaliser une condition sans
savoir a priori le nombre d’itérations nécessaires pour y parvenir.

3.1 Boucle for ... to ... end
La boucle itérative for sert à exécuter une suite d’instruction un nombre prédé-
fini de fois. Ci-contre sa syntaxe générale :

for variable in valeur_debut:pas:valeur_fin
instruction 1
instruction 2
. . .

end

La variable s’appelle aussi compteur. Les valeurs successives de cette variable
peuvent être utilisées au fil de la boucle.

Exemples

Tapez les deux exemples ci-dessous dans Scinotes :

a=2
for i=1:1:10
a=a+3

end

Dans le premier cas, i a été utilisé purement pour compter. A chaque nouvelle
valeur de i, a est augmenté de 3.

a=2
for i=3:2:10

a=a+i
end

Dans le second exemple, i part de 3 et prend toutes les valeurs de 2 en 2 jusqu’à
dépasser 10. Chacune de ces caleurs est tour à tour ajoutée à a
Autres exemples : Vous verrez mieux le fonctionnement de l’algorithme en pla-
çant l’affiche disp dans la boucle juste avant le end, pour voir comment les valeurs
évoluent.

1) Table de multiplication

x=[] // crée un vecteur vide
table=input('table de ? ')
for i=1:1:10

x=[x,i*table];
end
disp(x)

2) Quelle somme ce programme calcule-t-il ?

n=input("Entrer un nombre entier : ");
s=0;
for i=1:2:(2*n+1)

s=s+1/i;
end
disp(s)

Le premier exemple calcul tous les résultats des tables de multiplication et les
stocke dans un vecteur qui se construit au fur et à mesure. On pourrait enjoliver
le résultat en affichant l’opération suivie du résultat.

Le second programme calcule
n
∑
k=1

1
2k+1 . On aurait pu écrire la boucle de manière

plus intuitive avec :
n=input("Entrer un nombre entier : ");
s=0;
for k=1:1:n

s=s+1/(2*k+1)i;
end
disp(s)

La version proposée est plus rapide (en termes de nombre d’opérations) et plus
dans l’« esprit » Scilab .

À retenir…

⋆ Noter que Scinotes indente automatiquement les commandes entre for et end. Par souci de
clarté, veillez à respecter cette convention dans la rédactions à la main de vos algorithmes.

⋆ Retenir la syntaxe x=[x,i*tab] qui crée un vecteur avec le contenu de x « augmenté » de
la nouvelle valeur

⋆ n’hésitez pas à supprimer le ; et refaire tourner le programme pour voir les étapes de
construction du vecteur affiché en sortie.

⋆ la commande sum([1:2:2*n+1].^(-1)] permet de remplacer la boucle for pour ce genre
de cas. Vous pouvez tester sur une ou deux valeurs.

⋆ Dans le premier algorithme, on peut remplacer 1:1:9 par 1:9. Je conseille d’utiliser systé-
matiquement la syntaxe avec pas pour ne pas en oublier la tournure.

⋆ Il est possible (et souvent utile) d’imbriquer plusieurs boucle for

Tableau 3.1 :

Exercices

Exercice 20
Pour chaque algorithme ci-dessous, décrire au brouillon les étapes de l’algorithme,

▷ Pour comprendre
un algorithme, il faut
se mettre à la place de
l’ordinateur. Suivez les
instructions dans l’ordre
et voyez comment les
valeurs prises par les
différentes variables
évoluent au fil des com-
mandes. Pour cela, on
utilise en général un ta-
bleau comportant une
colonne par variable, et
une ligne par cycle.

puis l’affichage en sortie puis vérifier dans la console ou dans Scinotes .

x=4 ;
for i=1:x

y=0 ;
y=y+i

;
end
y

x=4 ;
y=0 ;
for i=1:x

y=y+i
;
end
y

a=5 ;
for i=1:3

a=a+1
;

x=a ;
end
x

a=5 ;
for i=1:3

a=a+1
;
end
x=a ;
x

a=5 ;
x=a ;
for i=1:3

a=a+1
;
end
x

Exercice 21
Écrire un programme qui demande à l’utilisateur un entier n et affiche en sortie
le nombre 0, 9n × 2. Ce programme ne devra pas faire usage de la commande ^

Exercice 22
Voici un algorithme :

u=0
for i=1:6

u=2*u+3
disp(u)

end

1. Reconnaître la suite (un) dont on calcule les termes et préciser le but de
l’algorithme

2. Modifier l’algorithme pour qu’il calcule u100

Exercice 23
Écrire un programme qui demande une valeur de n et calcule n!.

Si vous êtes en avance, améliorez votre code pour que le programme renvoie 1 pour
0 ! et un message d’erreur si n est négatif ou non entier.

Exercice 24 Une somme

Écrire un programme qui calcule la somme Sn = ∑
n
i=0 sin(i) à l’aide d’une

boucle for.

Si vous êtres en avance…

⋆ Saurez-vous retrouver la commande sans boucle faisant intervenir les vec-
teurs ?

⋆ Modifiez le programme précédent pour qu’il renvoie en sortie un vecteur
[min;max] où min (resp. max) est la plus petite (resp. grande) valeur prise par
Sn, pour n ∈ [[1; n]]

Exercice 25 Les tables, la suite

Écrire un programme qui affiche 10 vecteurs ligne contenant les résultats des
tables de multiplication des nombres de 1 à 10.

Si vous êtes en avance, saurez-vous modifier l’algorithme pour qu’il présente de belles
tables de multiplication comme au dos des cahiers de votre enfance ?

Exercice 26 une suite…

(un) est la suite définie par u0 = 0 et un+1 = √un + 1.

Écrire un programme qui calcule u10. Le modifier pour qu’il demande une valeur
n et crée un vecteur colonne qui contiennent les n premiers termes de la suite.
Que remarque-t-on ? Modifier u0, faire des tests et émettre une conjecture sur
(un).

Exercice 27 suite de Fibonacci

La suite de Fibonacci est définie par :

{
u0 = u1 = 1
un+2 = un+1 + un

Écrire un programme qui demande une valeur n à l’utilisateur et calcul un.

Faire des tests et conjecturer le comportement de la suite quand n devient grand.
Modifier ensuite légèrement le programme pour qu’il affiche également les termes
de la suite (

un
un−1)n≥2

et en conjecturer le comportement à l’infini.

3.2 Boucle while ... condition
La boucle while, ou itération conditionnelle sert à répéter des commandes jus-
qu’à ce qu’une certaine condition soit réalisée. De même que dans les structures
conditionnelles, la condition est un booléen.

while (booleen)
instruction 1
instruction 2
. . .

end

Remarques

⋆ On peut aussi ne jamais entrer dans la boucle si, dès le départ la condition
n’est pas remplie.

⋆ Il faut prendre garde au fait que cette structure peut donner lieu à une itéra-
tion sans fin. En effet, si la condition n’est jamais remplie, le programme ne
s’arrêtera jamais. Il faut donc toujours s’assurer que le programme se termine.
Pour cela, on peut :

− démontrer mathématiquement que la condition finira par être remplie
− implanter un compteur, qui correspond à un nombre maximal autorisé

d’itérations.
On peut même envisager les deux au cas où le nombre d’itérations nécessaire,
bien que théoriquement fini, soit un très très grand nombre.

⋆ Si le programme ne s’arrête pas, ou pas assez vite, on peut en interrompre
le déroulement. En effet, les menus restent accessibles et on trouvera la com-
mande abandonner dans le menu Contrôle

Exemples :

On peut démontrer que la suite définie par u0 = 1 et ∀n ∈ ℕ, un+1 = 1 − eun

converge vers 0.
On souhaite trouver la plus petite valeur de n pour laquelle on a |un| < 10−2.

Le programme suivant répond à la question.

x=-1;
n=0;
while abs(x)>10^(-2)

x=1-exp(x);
n=n+1;

end
disp(n)

si vous lancez plusieurs
fois le programme,
vous obtiendrez une
fluctuation du temps
de calcul. Peut-être
l’occasion de faire des
statistiques plus tard
dans l’année…

Vous pouvez le taper dans Scinotes et constater qu’il faut 59687 itérations pour
aboutir à la précision souhaitée. Notez que Scilab s’occupe de ces quelques calculs
en 0,559s sur mon ordinateur. Si vous voulez savoir combien de temps met le votre,

ajoutez tic() juste avant la boucle while et t=toc() juste après. La variable t
contient alors le temps d’exécution.

Remplacez maintenant 10^(-2) par 10^(-3) dans l’algorithme et exécutez le. Le
temps nécessaire est un peu plus conséquent. 49,804s chez moi pour effectuer les
5996985 opérations !

x=-1;
n=0;
tic();
while abs(x)>10^(-2)

x=1-exp(x);
n=n+1;

end
t=toc()
disp("n = "+string(n))
disp("calculé en "+string(t)+"s")

Remarque

Dans l’exemple ci-dessus, la syntaxe de disp est légèrement différente de celle vue
au TP précédent. Ici, on concatène les éléments de texte après avoir transformé la
variable numérique t en texte (chaîne de caractères) via la commande string().
Cette syntaxe présente deux avantages :

⋆ Les éléments du texte sont affichés dans l’ordre d’écriture
⋆ la valeur est intégrée à la phrase (et non affichée en dessous)

3.3 Exercices

Exercice 28 deux dés

Nous verrons dans un
prochain TP que l’on
peut aussi simuler un
dé avec une variable
aléatoire suivant une
loi uniforme sur [[1; 6]],
les principales loi de
probabilités sont déjà
implémentées dans
Scilab

Écrire un programme qui simule le lancer de deux dés à six faces, affiche le résultat,
et répète l’opération jusqu’à obtention d’un double.

Rappel : On peut simuler un dé avec 1+floor(6*rand())

Exercice 29 mission

Votre programme, si vous l’acceptez, s’appellera bond (j’aime bond). Un nombre
3.1x est augmenté ou
diminué de 1

x = 0 bondira de +1 ou -1 1 avec la même probabilité.
La mission du programme est de compter le nombre n de tirage nécessaires avant
que |x| > 007

si vous êtes en avance, lancez vous dans cette mission bonus : Votre programme
demandera un nombre p de répétition de l’expérience, stockera les valeurs n obtenues
successivement dans un vecteur afin de calculer le nombremoyen de bonds nécessaires.

Q met à votre disposition le rappel suivant :

▷ v=[]pour créer un vecteur vide
▷ v=[v;n]pour ajouter la valeur n au vecteur
▷ length(v)donne le nombre de termes de v

Quand votre programme est fonctionnel, vous pouvez alors tester jusqu’à p = 10000
pour avoir une bonne approximation du nombre moyen de tirages nécessaires, et
observer comme le résultat se stabilise avec l’augmentation des valeurs de p

Exercice 30 Devinette

Écrire un programme qui choisit au hasard un entier entre 1 et 100. Ensuite, le
programme demande à l’utilisateur de deviner quel nombre a été choisi et répond
« trop grand » ou « trop petit » selon la valeur proposée.
Le programme propose jusqu’à ce que l’utilisateur ait trouvé.

Si vous êtes en avance, modifiez le programme pour qu’il traque les incohérences
stratégiques du joueur en affichant le message : « on a déjà dit que c’était plus
grand/petit que <valeur> ». Pour cela, le programme comportera une variable max
(resp. min) gardant une mémoire de la plus grande (resp. petite) valeur proposée
antérieurement et comparant à chaque tirage.

Exercice 31 Fibonacci

Reprendre le programme calculant les termes de la suite de Fibonacci. En co-
pier/coller le texte dans une nouvelle fenêtre de Scinotes .

1. Adapter le code pour qu’il demande un entier p (exposant de 10p) puis qu’il
détermine le premier entier n à partir duquel ϕn dépasse 10p

2. Copier/coller le da Fibonacci code dans une autre fenêtre . Sachant que la suite

(un)n∈ℕ définie par un = ϕn+1
ϕn

converge vers le nombre d’or Φ = 1+√5
2 ,

déterminer le plus petit entier n tel que |un − Φ| < 10−p pour p demandé
par une invite.

3.4 Quelques corrigés :

Exercices sur la boucle for

// toutes les tables de 1 à 10 /////////////////////////////
disp("Tables de 1 à 10")
for table=1:1:10

x=[]
for i=1:1:10

x=[x,i*table];
end
disp(x)

end

Tableau 3.2 :

Ici le programme crée un vecteur qu’il remplit pour chaque table, affiche quand il
a finit puis l’écrase et recommence. L’affichage est amélioré en créant un tableau
comme dans le script suivant :

//TP3 //////////////////////////////////////////////////
// toutes les tables de 1 à 10 /////////////
disp("Tables de 1 à 10")
T=[]
for table=1:1:10

x=[];
for i=1:1:10

x=[x,i*table];
end
T=[T;x];

end
disp(T)

Tableau 3.3 :

//TP3 //////////////////////////////////////////////////
// tables de 1 à 10 - version jolie //////////
disp("Tables de 1 à 10")
for t=1:1:10

c=""
for i=1:1:10

d=string(t)+"*"+string(i)+"="+string(t*i)
c=c+d+" ";

end
disp(c)
disp("____________")

end

Tableau 3.4 :

//TP3 //////////////////////////////////////////////////
u=0;
for i=1:1:10

u=sqrt(u+1);
end
disp(u)

Tableau 3.5 :

// TP3 //////////////////////////////////////////////////
//une suite
n=input("entrez un entier n : ")
u=5;x=[u];
for i=1:1:n

u=sqrt(u+1);
x=[x;u];

end
disp(x)

Tableau 3.6 :

// TP3 //////////////////////////////////////////////////
// Suite de Fibonacci
n=input("donner une valeur de n : ")
u=1;v=1; //u représentera le un courant et v le terme précédent
if n>=2 then

for i=2:1:n // on commence à 2 pour que le premier calcul soit celui de u2
// et le dernier soit bien celui de un

a=u; //on met en mémoire la valeur courante
u=u+v; //on calcule la nouvelle valeur
v=a; //le terme précédent prend la valeur

//de l'ancienne valeur courante
// disp("u"+string(n)+" = "+string(u))
// décommenter la ligne ci-dessus pour afficher tous les termes de la suite.

end
disp(u)

else
if n<0 then

disp("on a dit positif !")
else

u=1; //il est préférable de quand même donner une valeur à u
disp(u)

end
end

Tableau 3.7 :

Exercices sur la boucle while

// TP3 - Lancer de deux dés /////////////////////////////////////////////
// jusqu'à obtention d'un double
//
de=0 // on fixe des valeurs
DE=1 // différentes pour entrer dans la boucle
n=0 // mise à zero du compteur
while de<>DE

de=1+floor(6*rand())
DE=1+floor(6*rand())
n=n+1
disp("lancer n°"+string(n)+" : "+string(de)+ " et "+string(DE))

end
disp("Vous avez obtenu une paire !")

Tableau 3.8 :

// TP3 //////////////////////////////////////////////////
// Bond 007 //////////////////////////////////////////////
x=0
n=0
while abs(x)<007

u=1-2*floor(2*rand())
x=x+u
n=n+1

end
disp("il a fallu "+string(n)+" bonds")

Tableau 3.9 :

// TP3 //////////////////////////////////////////////////
// Bond 007 ////////////////////////////////////////////
// avec nombre moyen de bonds nécessaires //////////////
v=[]
for i=1:1:10000

x=0
n=0
while abs(x)<007

u=1-2*floor(2*rand())
x=x+u
n=n+1

end
v=[v;n]

end

m=sum(v)/length(v)
disp("bonds",m,"en moyenne, il faut :")

Tableau 3.10 :

// TP3 - devine le nombre /////// version simple //////////////
//
nombre=1+floor(100*rand())
test=0
while test<>nombre

test=input("Devine le nombre ! ")
if test<nombre then

disp("trop petit ! essaie encore :-)")
else

if test>nombre then
// le message qui suit ne convient pas si test=nombre donc
// il faut un else...if

disp("trop grand ! essaie encore :-)")
end

end

end
disp("bravo, tu as trouvé !")

Tableau 3.11 :

// TP3 - devine le nombre /////////////////////////////////////////////////////
// /////////////////// Version avec mémoire ////////////////////
nombre=1+floor(100*rand())
test=0
max=101;min=0
while test<>nombre

test=input("Devine le nombre ! ")
if test<nombre then

if test<=min then
disp("on a déjà dit que le nombre était plus grand que "+string(min))

else
min=test

end
disp("trop petit ! essaie encore :-)")

else
if test>nombre then

if test>=max then
disp("on a déjà dit que le nombre était plus petit que "+string(max))

else
max=test

end
disp("trop grand ! essaie encore :-)")

end
end

end
disp("bravo, tu as trouvé !")

Tableau 3.12 :

// TP3 //////////////////////////////////////////////////
// Suite de Fibonacci /////// infini ///////////////////
p=input("entrez l''exposant p de 10^p : " )
u=1;v=1;
n=0
while u<10^p

a=u;
u=u+v;
v=a;
n=n+1;

// disp("u"+string(n)+" = "+string(u))
end

disp("n = "+string(n))

Tableau 3.13 :

TP4 | Matrices & vecteurs

Nous avons déjà vu que, pour Scilab , tout est vecteur, ou plutôt matrice. L’objet
de ce TP sera de parcourir les principales commandes relatives au calcul matriciel
et vectoriel. Nous en avons déjà vu plusieurs, qui seront reprises dans ce TP.

4.1 Vecteurs 34
4.2 Matrices 38
4.3 Résolution de systèmes linéaires 41
4.4 Approfondissements 42
4.5 Un peu de culture G : Alan Turing et la pomme d’Apple 43
4.6 Quelques corrigés 46

4.1 Vecteurs

Exemples

Quelques exemples sont plus parlants que des commandes théoriques alors tapez
les commandes suivantes dans la console. Vous pourrez compléter à droite pour
vous souvenir de l’effet de chacune d’elles.

Commençons par quelques exemples avec des vecteurs, pour la plupart déjà ren-
contrés

x=1:5

x=1:2:10

y=x'

x=3:-1:-2

y=1:-1:9

z=[1;50;0.2;0]

z=[1,50,0.2,0]

z=[1,50,0.2,0]'

x=[]
x=[x,1]
x=[x,2]
x=[x,3]
for i=4:10

x=[x,i]
end

Reprendre les mêmes commandes avec un ; (en console)

A retenir :

⋆ on définit un vecteur vide avec []
⋆ on peut définir un vecteur par ses composantes séparées par , ou ; (ligne

ou colonne)
⋆ , continue la ligne et ; passe à la ligne suivante (cf matrices)
⋆ un vecteur peut être complété au fur et à mesure avec

u=[u,complement]
⋆ ceci est bien pratique pour stocker des valeurs en vue de les représenter

graphiquement ou y faire référence plus tard.

Remarque avancée (hors programme)

Quand on écrit une boucle for, comme for i=1:10 la variable i prend sucessive-
ment la valeur de chaque composante du vecteur 1:10. Ainsi, si u est un vecteur,
on peut définir une boucle for i=u . . . instructions . . . end où la
variable numérique i prendra successivement les valeurs des composantes de u.

Encore une commande vectorielle utile

linspace(a,b,k) crée un vecteur comportant k valeurs (y compris a et b éga-
lement réparties dans [a; b]. Ainsi, linspace(a,b,k) a le même effet que la
commande a:(b-a)/(k-1):b. En effet k valeurs équiréparties entre a et b défi-
nissent (k − 1) intervalles.

Tapez en console :

linspace(0,10,10)
linspace(1,10,10)
linspace(0,1,10)
1:0.1:10
linspace(0,1,11)

Ces quelques exemplesmontrent à quel point il faut faire attention à la signification
de chaque paramètre. La commande n’est pas indispensable à retenir mais peut
s’avérer commode pour certaines manipulation. Nous en reparleront dans le TP
sur l’intégration.

Exemples de calcul avec les vecteurs

Ici encore de la redite pour rafraîchir la mémoire. Vous pourrez compléter le
tableau avec la description de l’action de chaque commande ci-dessous.

x=1:9
sum(x)
prod(x)

length(x)

2*x

x.^2

exp(x)

rand(1,10)
u=floor(2*rand(1,10))
v=floor(2*rand(1,10))
u==v

Les opérations que l’on fait habituellement avec un nombre, Scilab applique en
blocs à chaque composante d’un vecteur.

Exercice 32 : Table de valeurs
On utiliser ceci pour obtenir un tableau de valeurs rapide, par exemple pour tracer
une courbe. Intéressons-nous à la fonction « sinus ».

attention !
-2%pi:0.1:2%pi ne
fonctionne pas,
pensez aux *

⋆ Créer un vecteur x dont les composantes parcourent l’intervalle [−2π; 2π]
à partir de −2π de 0,1 en 0,1 puis un vecteur y contenant les images des
composantes de x par la fonction sinus.

⋆ Vous pouvez ensuite afficher avec plot(x,y)

Si vous êtes en avance, allez faire un tour du côté de l’aide avec help plot où
vous trouverez un tas d’exemples. Mais nous reviendrons plus en détails sur cette
commande plus tard.

Rappel : Pour effacer la fenêtre graphique entre deux essais, tapez clf()

4.2 Matrices
Les matrices se définissent essentiellement de la même manière que les vecteurs.

Par exemple, vous pouvez concaténer des vecteurs, ou des matrices :

x=1:5
M=[x;x]
M=[M;M]
M=[M,M]
M=[M;2*M]
x=1:10;M=[];
for i=1:10

M=[M;i*x]
end

Si vous voulez utiliser Scilab pour réfléchir à un exercice comportant des matrices,
vous pouvez aussi définir les matrice par leurs coefficients. Il suffit de les rentrer
ligne par ligne.

Les coefficients d’une même ligne sont séparés par , et ; change de ligne

Par exemple :

M=⎛
⎝

1 2 3
4 5 6
7 8 9

⎞
⎠

M=[1,2,3;4,5,6;7,8,9]

Cette syntaxe est orientée ligne (dans le sens matrice d’un système). Si vous
souhaitez définir la matrice par ses vecteurs colonnes (dans le sens image d’une
base), vous pouvez avoir recours à l’astuce suivante :

Si les colonnes sont⎛
⎝

1
2
3
⎞
⎠

,

⎛
⎝

2
4
6
⎞
⎠

et⎛
⎝

0
1
0
⎞
⎠

M=[1,2,3;2,4,6;0,1,0]'

Tout est dans le ' qui renvoie la transposée de la matrice comme nous le rappelle-
rons tout à l’heure.

Exemples de calcul avec les vecteurs

M
M'
rank(M)

size(M)

x=diag(M)
N=diag(1:5)
N=diag([1;-1;2;-2;3])
M
diag(diag(M))
M=[1,2,0;0,2,1;3,5,1]
inv(M)
M^(-1)
M^.(-1)
M*inv(M)
x=5:7
x'
M*x
M*x'
x'*M
A=[1,2,3;4,5,6]
B=[1,2;3,4;5,6]
A*B
B*A
x*x'
x'*x
inv(A)
M
M(1,2)
M(2,1)
M(2,1)=0
M^2
M.^2
M+1
M-2
min(M)
max(M)

Toutes les opérations « coefficient par coefficient » vues sur les vecteurs fonc-
tionnent également avec les matrices.

Attention toutefois. On peut définir l’exponentielle de matrice, qui n’est pas la
matrice dont les coefficients sont les exponentielles des coefficients de la matrice
de départ, une racine carrée de matrice… Vous pouvez tester ces opérateurs avec
exp(M), sqrt(M)…

Faites bien attention que M^2 calcule le carré de la matrice, au sens de la multiplica-
tion de matrice, alors que M.^2 calcule le carré de chaque coefficient. De manière
générale, le . devant une opération opère coefficient par coefficient.

Plus de définitions de matrices

Essayez aussi ces commandes et notez leur effet à côté

zeros(4,6)
zeros(6,4)
ones(4,6)

eye(4,6)
eye(6,4)
eye(3;3)
rand(4,6)
rand(4,6)
floor(2*rand(4,6))
floor(10*rand(4,6))

Pour vous souvenir de eye, pensez à la prononciation anglaise du I qui désigne la
matrice identité

Exercice 33
Coder chacune des matrices suivantes dans Scilab.

A=⎛
⎝

0 1 1
1 0 1
1 1 0

⎞
⎠

, B=⎛
⎝

1 3 3
3 −1 3
3 3 2

⎞
⎠

, C=⎛
⎝

1 2 3
2 4 6
3 6 9

⎞
⎠

, D=⎛⎜⎜
⎝

0 0 1 1
0 0 1 1
1 1 0 0
1 1 0 0

⎞⎟⎟
⎠

Exercice 34
Coder en Scilab la matrice

N=⎛⎜⎜
⎝

0 1 1 1
0 0 1 1
0 0 0 1
0 0 0 0

⎞⎟⎟
⎠

Puis calculer N2, N3, N4,

Extraction, sous-matrices

Essayez aussi ces commandes et notez leur effet à côté
M=[1:5]'*[3:10]
M(1,:)
M(2,:)
M(:,1)
M(2,:)
M(:,1)=[1:5]'
M(1,:)=[1:8]
v=[1,2,5]
M(v,:)
M(:,v)
M(v,v)

4.3 Résolution de systèmes linéaires
L’étude de problèmes mathématiques amène souvent à résoudre des systèmes
d’équations de la forme :

⎧
⎨
⎩

a11x + b12y + c13 = b1
a21x + b22y + c23 = b2
a31x + b32y + c33 = b3

En posant A = ⎛
⎝

a11 a12 a13
a21 a22 a23
a31 a32 a33

⎞
⎠

, X = ⎛
⎝

x
y
z
⎞
⎠

et B = ⎛
⎝

b1
b2
b3
⎞
⎠

,

le système peut s’écrire AX = B

Si A est inversible, on peut le résoudre en multipliant à gauche par l’inverse A−1

de A.
On trouve X = A−1B qui se code en Scilab x=inv(A)*b

Cette méthode suffira dans la plupart des cas mais la méthode « propre » à Sci-
lab utilise la syntaxe x=A\b, appelée division à gauche.
C’est « comme si » on multipliait à gauche par A−1. L’intérêt de cette syntaxe
est qu’elle renvoie une solution (la meilleure au sens des moindre carrés) même si
la matrice A n’est pas inversible, et en particulier si elle n’est pas carrée.

Néanmoins, dans les cas où la résolution théorique aboutirait à une infinité de
solutions, Scilab ne les détermine pas toutes

Exercice 35 Un système…

Utiliser Scilab pour résoudre le système suivant :

⎧
⎨
⎩

3x − 2y + 5z = 27
x − 3y + z = 11
2x + 3y + 7z = 24

Exercice 36

1. Quelle est la commande Scilab qui permet de savoir si une matrice est inver-
sible ?

2. Créer une dans Scinotes une matrice 3 × 3 alétoire dont les coefficients
peuvent valoir 0 ou 1 avec la même probabilité. Tester si elle est inversible.

3. Écrire un programme qui détermine une valeur approchée de la probabilité
qu’une matrice dont les coefficients suivent la loi 𝒰({0; 1}) uniforme sur {0; 1}
(c’est à dire qui vaut 0 ou 1 avec la même probabilité) soit inversible.

4. Si vous êtes en avance, modifiez le programme pour tester si la probabilité est
la même avec une matrice 4×4, puis avec une matrice 3×3 dont les coefficients
suivent 𝒰([[0; 9]])

4.4 Approfondissements

Échantillonnage, tests multiples

On peut envisager la commande rand(n,m) comme un moyen de générer une
matrice aléatoire dont chaque composante suit une loi uniforme sur [0; 1].
On peut aussi voir cette matrice comme m échantillons de n répétitions d’une
variable aléatoire. Essayez la commande floor(10*rand(10,6))

Maintenant, essayez floor(10*rand(1,10));, puis x==6, x==1, x>3, x<=6…
Un test appliqué à un vecteur renvoie un vecteur correspondant aux résultats des
tests de chaque coordonnée.
Pour compter le nombre de résultats égaux à 1, on peut définir un vecteur v=[x==1]
puis en ajouter les composantes avec sum(v). Les comptent pour 1 (comme en
calcul binaire) et compte pour 0.

Exercice 37
Créer un vecteur x et un vecteur y comportant chacun 100 composantes aléatoires
suivant chacune la loi 𝒰([[0; 9]])
Ensuite écrivez une commande qui compte le nombre de composantes de x supé-
rieures ou égales à 5. Comptez ensuite les composantes de x et y qui sont égales
(c.a.d le nombre d’indices i tels que xi = yi). Enfin, le nombre de composantes de
x strictement plus petites que la composante de y correspondante.

Exercice 38 On rappelle que la loi binômiale ℬ(𝓃 ; 𝓅) compte le nombre de
succès obtenus en répétant n fois une épreuve à deux issues dont la probabilité
de succès est p.

1. Écrire dans Scilab une commande qui crée un vecteur dont les composantes
simulent la répétition de 10 épreuves à deux issues équiprobables. Compléter
cette commande pour compter le nombre de succès

2. Écrire en Scilab un programme qui répète 100 fois ce qui précède et en conserve
les résultats successifs dans un vecteur. Afficher le résultat avec histplot(10,x)



TP4 | Matrices & vecteurs

43 Jean-Marc BERROU - 2014-2015

4.5 Un peu de culture G : Alan Turing et la pomme
d’Apple

Un texte extrait du site tatoufaux.com

En décembre 2013, la reine d’Angle-
terre accordait à titre posthume la
grâce royale à un mathématicien an-
glais, près de 60 ans après son dé-
cès. Condamné pour homosexualité
dans les années 50, cette réhabilita-
tion très réclamée reconnaissait im-
plicitement l’absurdité de cette loi in-
digne et d’autant plus cruelle qu’elle
avait probablement mené le « cou-
pable » au suicide, alors même qu’on
estime qu’il a personnellement per-
mis, grâce à son travail, que soient
épargnées des milliers de vies hu-
maines.

Figure 4.1 : Alan Turing

Au-delà de sa propre postérité, le
nom d’Alan Turing est également as-
socié à une société qui, plus que n’im-
porte quelle autre, suscite passions et
controverses : Apple.

Et plus particulièrement son logo, re-
connaissable entre mille et qui re-
présente une pomme croquée. Apple

n’est pas qu’une marque d’ordina-
teur, c’est aussi un phénomène de so-
ciété qui défie parfois toute logique.
Le mythe créé autour de son créateur
emblématique, idolâtré comme une
rock-star par ses fans, ou les queues
interminables devant les magasins
des semaines avant la sortie d’un
nouveau modèle de téléphone ou
d’ordinateurs ne sont que quelques
exemples de la place qu’a pris cette
marque dans la culture populaire.

Rien de surprenant donc à ce que
l’origine de ce logo fasse l’objet de
théories plus ou moins exotiques,
mais l’une d’elle particulièrement
alambiquée reste relativement te-
nace : celle de l’hommage à Alan Tu-
ring qui a d’ailleurs au moins le mé-
rite d’être poétique.

Revenons un petit peu en arrière.
Nous sommes en 1954, soit une
vingtaine d’années avant la création
d’Apple. Le génial mathématicien an-
glais décède dans des conditions so
romantic ! Condamné deux ans aupa-
ravant à la castration chimique après
que son homosexualité ait été révé-
lée (un délit en Grande-Bretagne jus-
qu’en 1967), Alan Turing meurt en
croquant une pomme empoisonnée

4.1 II n’existe pas de
preuves formelles car la
pomme retrouvée cro-
quée à côté de son corps
n’a jamais été analysée
mais Alan Turing était
un inconditionnel de
Blanche Neige et cette
version de sa mort est
très largement consi-
dérée comme la plus
plausible.

au cyanure1.

Une fin tragique pour ce scientifique
de haut vol qui a notamment mis au
point les algorithmes permettant de
décoder les messages de l’armée al-
lemande, et en particuliers ceux ré-
putés inviolables de la fameuse ma-
chine Enigma, jouant ainsi un rôle
déterminant dans la victoire des Al-

http://tatoufaux.com/?Le-logo-d-Apple-est-un-hommage-a

liés.

Une pomme croquée, un mathémati-
cien adulé, une fin de conte de fées,
il n’en fallait pas plus pour que l’on
voit dans cette pomme croquée de la
firme Apple, un hommage des pères
du Macintosh à l’un des pères de
l’informatique moderne. Pour autant
que l’on sache, il n’en est rien. Et
cette explication est d’ailleurs rela-
tivement fantaisiste. L’hommage à
Alan Turing pourrait se concevoir
mais utiliser un signe de mort et
d’empoisonnement pour une firme
fabriquant des ordinateurs serait as-
sez curieux. Le logo de 1976 est
composé de six bandes de couleurs
pouvant facilement être confondues
avec le drapeau arc-en-ciel, mais en
1976 ces couleurs n’évoquaient pas
comme aujourd’hui la communau-
té gay et d’ailleurs aucun des fon-
dateurs d’Apple n’était gay, ce qui
exclut un clin d’œil communautaire.
D’autant que si l’on regarde bien, ce
ne sont pas les mêmes couleurs.

À moins que l’origine du logo n’ait
pas d’histoire (Steve Job mangeait
des pommes, il décide d’appeler sa
boite Apple, et rien de plus), le
plus probable reste que la fameuse
pomme soit celle de Newton, autre
génie des sciences dont on dit qu’il
aurait compris les lois de la gravi-
té en en prenant une sur le coin du

crâne. Ce rapport à Newton est fa-
cile à vérifier, le premier logo d’Apple
computer compagnie le représentait
adossé à un arbre, une pomme au-
dessus de la tête. Difficile de faire
moins allégorique.

Mais alors pourquoi une pomme cro-
quée ?

Selon toute vraisemblance et de
l’avis même de Rob Janoff, le gra-
phiste qui a créé ce logo, c’était tout
simplement pour éviter qu’elle soit
confondue avec une tomate. Déce-
vant, certes, mais probablement au-
thentique.

Compléments au texte (inspirés de Wikipedia)

Après la guerre, il travaille sur un
des tout premiers ordinateurs, puis
contribue de manière provocatrice
au débat déjà houleux à cette pé-
riode sur la capacité des machines
à penser, en établissant le test de
Turing. Turing a mené également
des réflexions fondamentales réunis-
sant la science et la philosophie.
Dans l’article « Computing Machine-
ry and Intelligence » (Mind, octobre

1950), Turing explore le problème de
l’intelligence artificielle et propose
une expérience maintenant connue
sous le nom de test de Turing, où
il tente de définir un standard per-
mettant de qualifier une machine de
« consciente » ; Turing fait le « pa-
ri que d’ici cinquante ans, il n’y au-
ra plus moyen de distinguer les ré-
ponses données par un homme ou un
ordinateur, et ce sur n’importe quel

sujet. Il est en passe de gagner son pa-
ri quand on sait qu’en 2014, un ordi-
nateur a réussi à se faire passer pour
un adolescent dans un concours de
discussion entre humains et intelli-
gences artificielles. Vers la fin de sa
vie, il s’intéresse à des modèles de

morphogenèse du vivant conduisant
aux « structures de Turing ».
Sur le même thème, vous pourrez
(re)voir le biopic Imitation Game sur
Alan Turing et l’extrapolation de nos
réseaux sociaux dans le film Her.

4.6 Quelques corrigés
Pour l’exercice 2, on peut définir les matrices avec :

⋆ A=ones(3,3)-eye(3,3)

⋆ ou A=ones(3,3);diag(A)=[0,0,0];A

⋆ B=3*ones(3,3);B=B+(diag([1,-1,2]-3));B

⋆ attention, comparer à B=3*ones(3,3);B=B+(diag([1,-1,2])-3);B

⋆ u=[1,2,3];C=[u;2*u;3*u]

⋆ A=zeros(2,2);B=ones(2,2);D=[A,B;B,A]

Pour l’exercice 3,
N=[0,1,1,1;0,0,1,1;0,0,0,1;0,0,0,0],N^2,N^3,N^4

exercice 5

Essayer d’inverser une matrice ne permet pas de savoir si celle-ci est inversible. En
effet, si elle ne l’est pas, vous obtiendrez un message d’erreur. Une matrice carrée
est inversible ssi son rang est égal à son ordre.

Ici, la commande rank(M)==3 renvoie donc %T si elle est inversible et %F si elle
ne l’est pas.
M=floor(2*rand(3,3));rank(M)==3 crée une matrice aléatoire et teste son
inversibilité. Changez ; pour , si vous voulez voir la matrice.

On peut estimer la probabilité qu’une telle matrice soit inversible avec le pro-
gramme suivant :

// Estimation de la probabilité
// qu'une matrice aléatoire soit inversible
n=3 // dimension de la matrice
u=[]
N=input("Entrez un nombre entier : ")
for i=1:N

M=floor(2*rand(n,n))
u=[u,rank(M)==n] // on ajoute le résultat

// du test d'inversibilité
end
p=sum(u)/length(u) // évalue la probabilité
disp(p,"probabilité que la matrice soit inversible : ")

Le code ci-dessous est un peu plus général que ce qui était demandé et permet de
changer facilement la dimension de lamatrice. On peut remarquer avecN = 10000
itérations, que pourn = 3, la probabilité que lamatrice soir inversible est d’environ
0, 3. Elle est approximativement de 0, 7 pour n = 10 et et 1 pour n = 100.
On pourrait calculer « à la main » de manière exacte, la probabilité que la matrice
soit inversible pour les petites valeurs de n

prolongement

Le code qui représente graphiquement une estimation de la loi de la v.a.r X qui
à un entier n non nul associe la probabilité qu’une matrice dont les coefficients
suivent une loi uniforme sur {0; 1} soit inversible.

clf()
x=[]
for n=1:100 // 100 permet un temps de calcul raisonnable

u=[]
for i=1:100

M=floor(2*rand(n,n))
u=[u,rank(M)==n] // on ajoute le résultat

// du test d'inversibilité
end
x=[x,sum(u)/length(u)] // évalue la probabilité

end
plot(x)

Avec n = 100, le résultat est très approximatif. Si vous avez le temps (plusieurs
minutes), essayez avec n = 10000 et après une longue pause café, vous obtiendrez
la courbe ci-dessous, la probabilité qu’une matrice aléatoire soit inversible tend
rapidement vers 1avec la taille de celle-ci.

Les exercices sur les suites sont légion et particulièrement propices aux exercices d’algorithmique. L’objet
de ce TP sera de dresser un panorama des exercices d’algorithmique en lien avec les suites. Beaucoup de
redite au programme mais je recommande chaudement la maîtrise de ces codes pour les écrits à venir. Nous
ferons certainement un autre TP sur les suites plus tard en révisions.

Qui dit récurrence dit boucle itérative, les notions sont parallèles : l’une mathé-
matique, l’autre algorithmique. Nous allons donc énormément utiliser ces notions
après en avoir vu deux nouvelles :

5.1 Fonctions Scilab
Scilab est très riche en commandes et fonctions diverses. Toutefois, on peut pro-
grammer des fonctions personnelles, pour un usage spécifique et l’une des ques-
tions d’algorithmique les plus fréquentes est : « Écrire une fonction Scilab qui
fait ceci ou cela. »

Voyons la mise en œuvre sur un exemple. Tapez dans Scinotes :

function [y] = puissance(x,n)
// calcule la puissance n-ieme
y = 1 // valeur de y^0, et valeur initiale si n>0
if n > 0 then
for i = 1:n

y = y*x // on multiplie n fois x par lui-même
end

end
endfunction

Il est possible que Sci-
note propose l’exten-
sion .sci au lieu du
.sce habituel. C’est
normal. Il s’agit de l’ex-
tension propre aux
extensions de scilab (fi-
chiers internes quand
.sce est propre aux
programmes (externes).

Puis enregistrez sous le nom puissance. Quand on exécute le code, il se ne se
passe … rien.
Rien de visible en tout cas. En effet, l’exécution de ce code a ajouté à la liste des
fonctions une nouvelle. Si maintenant vous tapez en console :
--> puissance(2,3)
vous verrez que votre nouvelle fonction est disponible.

La syntaxe générale est la suivante :

function [<vecteurs sorties>] = nom_fonction(<entrées>)
<instructions>

endfunction

Les [ ] sont facultatifs pour une fonction dont la sortie est une variable unique.
Les crochets sont là pour rappeler que la sortie est un vecteur.

A retenir sur les fonctions

⋆ on commence par function et on termine par endfunction
⋆ Quand on lance une fonction, on la charge en mémoire mais aucun calcul

n’est fait
⋆ Pour utiliser une fonction, on l’appelle comme n’importe quelle autre com-

mande de Scilab .
⋆ Si la sortie comporte plusieurs nombres, plusieurs options sont possibles.

Soit on programme la fonction avec function u=fonction(x,y,z) et
u est évalué en tant que vecteur dans la fonction. Dans ce cas, la fonc-
tion renvoie ledit vecteur. Deuxième option, on implémente sous la forme
function [a,b]=fonction(x,y,z) et, par défaut, la fonction renvoie la
première variable a. Pour obtenir les deux, il faut appeler la fonction avec
une syntaxe du type : [u,v]=fonction(3,5). u contiendra alors la pre-
mière variable et v la seconde.

Exercice 39 Écrire une fonction Scilab qui prend en entrée un réel x et lui
associe e−x si x ≥ 0 et 0 dans tous les autres cas.

Exercice 40 Écrire une fonction Scilab qui prend en entrée un réel x et lui
associe x + 1 si x ∈ [−1; 0], −x + 1 si x ∈ [0; 1] et 0 dans tous les autres cas.

Exercice 41

1. Écrire une fonction sinxsurx qui, à un réel x non nul associe sin xx et associe
1 à 0. Testez votre nouvelle fonction.

2. Facultatif : si vous avez trouvé sans hésitation alors que tout le monde cherche
encore, sauriez prouver proprement que cette fonction est continue en 0

En cas de problème avec
le module graphique,
essayez la commande
usecanvas(%T)

3. Afficher le graphe de la fonction avec
x=[-10:0.1:10],plot(x,sinxsurx(x))

L’affichage du graphe nécessite que la fonction puisse travailler avec des vecteurs.
Or, si vous avez codé comme je l’imagine, la division par un vecteur n’est pas
définie.
La division par un vecteur, terme à terme s’obtient avec ./ au lieu de / qui concerne
les nombres uniquement.
Reste à régler le problème de la division par zéro. Pour cela, il suffit de remplacer
tous les zéros de x par ε, le %eps qui désigne le « zéro machine ». Par exemple,

x=max(abs(x),%eps) puis y=sin(x)./x) dans la fonction font l’affaire.
Prenez un temps pour vous en convaincre et voir pourquoi cette valeur absolue
ne change rien aux valeurs de la fonction.

Tracez ensuite la courbe de la fonction avec x=-5:0.1:5;plot(x,sinxsurx(x))

5.2 Récursivité

Définition

définition : Une procédure récursive est une procédure récursive…

Plus sérieusement, on parle de récursivité quand une fonction, dans sa définition,
s’appelle elle-même. Cette notion algorithmique est très liée à la notion de
récurrence en mathématiques.

Observons les deux exemples ci-dessous :

function [f]=fact(n)
// la factorielle en recursif

if n <= 1 then
f = 1

else
f = n*fact(n-1)

end

function [u]=fib(n)
// calcul du n ieme terme
// de la suite de Fibonnaci :
// fib(0) = 1, fib(1) = 1,
// fib(n+2) = fib(n+1) + fib(n)
if n <= 1 then

u = 1
else

u = fib(n-1) + fib(n-2)
end

Le premier exemple est classique et à retenir. Le second est un exemple classique
de mauvaise utilisation de la récursivité. En effet, l’itération classique (cf le TP
sur les boucles) est plus efficace, au sens qu’elle demande moins de calculs à la

On pourrait aussi le
vérifier avec tic() et
toc()

machine (les termes inférieurs de la suite sont calculés plusieurs fois).
Toutefois, la récursivité permet une implémentation simple et surtout, elle fonc-
tionne.
Aussi, si ça vous parle, n’hésitez pas à en faire usage car l’optimisation du code ne
fait pas partie des compétences requises en ECS. Nous ferons donc comme si les
calculs étaient instantanés et utiliserons la récursivité d’une manière parfois peu
recommandable mais … vous ne postulez pas à une école d’informatique !

Exemples

Une idée que l’on retrouve dans toutes les récursivités est que, pour connaître le
terme courant, on a besoin du précédent, qui nécessite le précédent, qui… Tous
ces précédents forment ce que l’on appelle la « pile » et il est important que la
fonction récursive contienne la valeur explicite du ou des premier(s) terme(s), le
bas de la pile.

Une fonction récursive contiendra donc :

⋆ Initialisation : un test pour déterminer si le cas demandé correspond à
la valeur initiale

⋆ Hérédité : un appel à elle-même au(x) rang(x) inférieurs n − 1, éven-
tuellement n − 2…

5.3 Exercices

Exercice 42 : You want to remember this one
Pour chacune des suites suivantes, écrire une fonction récursive qui prend un
entier n en entrée et renvoie un.

a. u0 = 1 et un+1 = √un + 5
b. u1 = 1 et un+1 = un + 1

un
c. u0 = 0 et un+1 = −1

2un − 3

Pour chaque suite, faire quelques tests afin de conjecturer la convergence éven-
tuelle et, le cas échéant, une valeur aprochée de la limite.

Travail à la maison : Déterminer, le cas échéant la valeur exacte l de la limite
et écrire une procédure qui demande un entier p et renvoie le premier indice n
tel que |un − l| < 10−p. Réfléchissez à la pertinence ou non de faire usage de la
fonction récursive écrite avant (même si ce n’est pas une compétence exigible).

Exercice 43 :
Écrire une fonction SCILAB récursive qui demande un nombre n en entrée et
renvoie la valeur de un pour la suite definie par :

⎧
⎨
⎩

u0 = 0
u1 = 1
un+1 = −3un + 2un−1

Remarquer par exemple
que Vn+1 = MVn

où Vn = (
un+1un ) et

M =
(
−3 2
1 0)

.

Si vous êtes en avance, écrivez un programme itératif qui fait le même travail.

Exercice 44 : Algorithmes de calcul des coefficients du binôme
Le but de cet exercice est d’écrire une fonction binome qui prend en entrée deux
entier n et p et renvoie (np).
Nous allons en fait en voir trois différents.

1. a. démontrer la formule
(

n
k)

= nk(

n − 1
k − 1)

b. utiliser cette relation pour écrire une fonction récursive de calcul des (nk)

Vous pourrez démar-
rer avec le vecteur
u=[1,1] et passer à
l’étape suivante avec la
scilabesque commande
: u=[u,0]+[0,u] que
vous pourrez tester en
console pour ensuite
l’implémenter dans une
fonction.

2. a. démontrer la fomule de Pascal :
(

n
k)

+
(

n
k + 1)

=
(

n + 1
k + 1)

b. utiliser cette formule pour écrire une fonction itérative de calcul des coeffi-
cients du binôme.

Si vous avez de l’avance, pour réviser les boucles, écrivez un programme ou une
fonction qui demande en entrée un nombren et renvoie unematriceM = (mij) ∈

ℳn,n+1(ℤ) telle que ∀i, j ∈ [[1; n]],mij =
(

i
j − 1)

Par exemple, pour n = 5, on obtiendrait la matrice :
Pensez à ce que vous
feriez par exemple dans
un tableur, et traduisez
le en code, en faisant
attention au nombre de
départ et de fin dans les
boucles !

⎛⎜⎜⎜⎜
⎝

1 1 0 0 0 0
1 2 1 0 0 0
1 3 3 1 0 0
1 4 6 4 1 0
1 5 10 10 5 1

⎞⎟⎟⎟⎟
⎠

On pourra commencer par créer une matrice de zéros aux bonnes dimensions,
remplir la première colonne de 1 et compléter les coefficients de la matrice ligne
par ligne grâce à la formule de Pascal. (cf triangle de Pascal)

Exercice 45 : Un classique !
On s’intéresse à la suite (un)n de premier terme u0 = 100 et vérifiant.

∀n ∈ ℕ, un+1 = √6 +√3un

.
Cette suite converge.
⋆ Écrivez une fonction fonc simulant x ↦ √6 +√3x

⋆ Écrivez une fonction suite qui calcule un pour tout n.

⋆ faites vous une idée de la limite puis écrivez un programme qui détermine le
premier entier n tel que |un − l| < 10−4

Exercice 46 ⋆⋆ : La suite de Syracuse est une suite d’entiers définie par un
premier entier u0 = N ∈ ℕ⋆ et la relation de récurrence

un+1 =
{

un
2 si un est pair

3un + 1 si unest impair

La conjecture de Syracuse affirme que, quel que soit le choix de u0, il existe un
rang p tel que up = 1. Le premier entier p vérifiant up = 1 est parfois appelé
« longueur de vol »

1. Écrire un programme Scilab qui demande un entier N en entrée, et renvoie
le plus petit entier pN tel que upN = 1 pour la suite de Syracuse de premier
terme u0 = N.
On pourra utiliser une fonction syr telle que un+1 = syr(un) puis une fonction
vol qui détermine la longueur de vol pour un entier N donné en entrée.

2. Adapter le programme précédent (que vous pourrez enregistrer sous un autre
nom) pour qu’il renvoie un vecteur x = (p1, ..., p500) dont la k-ième coordon-
née pk contient le rang du premier terme égal à 1 de la suite de Syracuse de
premier terme u0 = k

3. En déduire la plus grande longueur de vol, puis la moyenne des longueurs de
vol sur les entiers inférieurs à 500.

Figure 5.1 : Le nom de la suite vient de l’Université de Syracuse

Figure 5.2: Les surréa-
listes étaient amateurs
d’autoréférence

5.4 Culture G : L’autoréférence
"la poésie dit ce qu’elle dit en le disant ", Jacques Roubaud

L’autoréférence ne désigne pas que
la pratique de certains auteurs qui
placent leur bibliographie en… biblio-
graphie.

On la retrouve en mathématiques,
musique, arts, littérature… et elle a
de tous temps exercé une puissante
fascination, Narcisse ne le démenti-
rait pas. Universelle, archétypale, elle
est présente dans les motifs tradition-
nels Shipibos ouHopi d’Amérique du
sud, dans la musique Africaine, dans
les mandalas d’Inde, chez Zhuāng Zǐ
qui, dans la Chine du IV ̀me siècle, qui
rêve qu’il est un papillon, qui rêve
qu’il est un être humain… qui ne sait
plus qui il est.

En art, la symbolique du cercle et
de la spirale l’illustrent. L’œuvre de
M.C Escher y rend un impression-
nant hommage quand d’autres en
jouent de façon moins ostentatoire.

Pochette de l’album Ummagumma, de Pink
Floyd

En musique, le larsen de Jimi Hen-
drix est un son autoréférent produit
par l’enregistrement de lui-même, re-
produit déformé réenregistré, … jus-
qu’à saturation. J.S. Bach use de l’au-
toréférence dans la composition de
ses fugues qui se déroulent à l’infini,

sans parler de la musique comtempo-
raine de Philip Glass ou Steve Reich.

logo autoréférent de la Vache qui rit

La mise en abîme est un procédé
que l’on retrouve chez de nombreux
auteurs de littérature comme Mi-
lan Kundera, Georges Luis Borges,
Shakespeare, Pérec… ou dans des
œuvres cinematographiques comme
La nuit américaine, Inception, … Sartre
dans L’Être et le néant, disserte sur la
capacité de l’homme à être conscient
de lui-même et de ses actes puis de la
capacité prendre conscience du fait
d’être conscient. Philo toujours avec
Wittgenstein qui s’interroge :« Ce
livre ne sera peut-être compris que
par qui aura déjà pensé lui-même
les pensées qui s’y trouvent expri-
mées ».

En Mathématiques enfin, les frac-
tales font la part belle à l’autoré-
férence ainsi que le théorème d’in-
complétude de Kurt Gödel qui dé-
montre que quel que soit le sys-
tème d’axiomes choisi au départ, on
peut formuler une proposition indé-
cidable (on ne peut prouver sa vérité
ou fausseté).
L’autoréférence est à l’origine de

maints paradoxes qui font les dé-
lices de Geluck et de son Chat. Et
vous, sauriez-vous dire si la phrase

« Cette phrase est vraie » est vraie
ou fausse…

Pour prolonger le thème :

⋆ un article de JP Delahaye paru dans Pour la Science : http ://www.ecriture-
art.com/autoreference.pdf

⋆ Gödel, Escher, Bach, les brins d’une guirlande éternelle, Douglas Hofstadter.

⋆ œuvre de M.C Escher sur http ://www.mcescher.com/

⋆ Wikipédia, http ://fr.wikipedia.org/wiki/Mise_en_abyme

il maestro, M.C. Escher

http://www.ecriture-art.com/autoreference.pdf
http://www.ecriture-art.com/autoreference.pdf
http://www.mcescher.com/
http://fr.wikipedia.org/wiki/Mise_en_abyme

5.5 Quelques corrigés :

Fonctions

Exercice 2

function [y]=fonction(x)
y=max(-abs(x)+1,0)

endfunction

Pour que la fonction reste compatible avec un traitement vectoriel donc, par
exemple avec une utilisation graphique via
x=[-5:0.1:5];plot(x,fonction(x))
la syntaxe proposée ici est préférable à un test qui, appliqué à un vecteur ne sera
jamais vérifié.
Retenir l’usage de max() ou min() pour ce type d’usage vectoriel.

// sinxsurx ///////
// version naive
// ne fonctionne qu avec des nombres
function [y]=sinxsurx(x)
if x<>0 then

y=sin(x)/x
else

y=1
end
endfunction

// sinxsurx ///////
// version vecteurs
function [y] = test(x,n)
x=max(abs(x),%eps)
// %eps est suffisamment petit
// pour ne pas changer la valeur approchée
// mais surtout %eps est non nul !
y=sin(x)./x // avec ./ pour diviser terme à terme
endfunction

Exercice 4

function [u]=suite1(n)
if n==0 then

u=1
else

u=sqrt(suite1(n-1))+5
end

endfunction

function [u]=suite2(n)
if n==1 then

// Attention la suite commence pour n=1
u=1

else
u=suite2(n-1) //on "stocke"
// le résultat pour éviter deux appels
u=u+1/u //car on en a besoin deux fois

end
endfunction

function [u]=suite3(n)
if n==0 then

u=0
else

u=-0.5*suite3(n-1)-3
end

endfunction

Rappelez vous d’éviter d’appeler deux fois la procédure dans une récursivité. On
appelle une fois, on stocke la valeur que l’on utilise ensuite à notre gré.

Exercice 5

// suite récurrente linéaire d'ordre 2 ///////
// version récursive //////////////
function [u]=rec_lin(n) // version récursive

if n<=1 then
u=n // astuce pour cette suite uniquement

else
u=-3*reclin(n-1)+2*reclin(n-2)

end
endfunction

Bien que peu performante d’un point de vue complexité, cet algorithme a le mérite
d’être très simple et intelligible. Plus performante, la version itérative ci-dessous
qui ne réalise qu’une fois chaque calcul.

// suite récurrente linéaire d'ordre 2 ////
// version itérative ///////////
function [u]=rec_lin_it(n) // version itérative

M=[-3,2;1,0]
U=[1;0] // attention à bien créer une colonne
for i=1:n

U=M*U
end
u=U(2)

endfunction

Exercice 6 : Coefficients du binôme

// calcul des coefficients du binôme /////
// version récursive //////////////////////
function [C]=cbinom(n,k)

if k=0 then
C=1

else
C=(n/k)*cbinom(n-1,k-1)

end
endfunction

J’apprécie l’élégance de la version itérative suivante, bien qu’elle comporte quelques
pièges à retenir.

// calcul des coefficients du binôme //////
// version itérative //////////////
function [C]=cbinomit(n,k)

u=[1,1] // lors de l'initialisation n=1
for i=2:n // donc on commence à 2 ici

u=[0,u]+[u,0] // peu visible
// mais il s'agit bien dela formule de Pascal

end
C=u(k+1) // attention, le premier coefficient
//correspond à 0

endfunction

Enfin, mais c’est vraiment pour s’entraîner sur les boucles, un algorithme d’affi-
chage du triangle de Pascal dans une matrice.

n=input("entrez un nombre n : ")
M=zeros(n,n+1)
M(:,1)=1 // une première colonne de 1
M(1,2)=1 // le deuxième 1 pour

//commencer le triangle de Pascal
for i=2:n // on va remplir chaque ligne

for j=2:(i+1) // pas besoin d'aller
// au bout de la ligne

M(i,j)=M(i-1,j-1)+M(i-1,j)
end

end
disp(M)

Exercice 8 : Suite de Syracuse

// syracuse /////////
function [y]=syr(x)
if modulo(x,2)==0 then // commande la plus simple

// pour savoir si un nombre est pair
// on peut aussi utiliser x-2*floor(x/2)==0
// qui n'est vrai que si x est pair

y=x/2
else

y=3*x+1
end
endfunction

u=input("u_0= ")
k=0,
while u<>1

k=k+1
u=syr(u)

end
disp(u,"longueur de vol : ")

// syracuse 2 /////////
// on suppose implémentée la fonction syr
L=[]
for i=1:500

k=0
u=i
while u<>1

k=k+1
u=syr(u)

end
L=[L,k]

end
disp(max(L), "la plus grande longueur de vol est : ")
disp(sum(L)/length(L), "La longueur de vol moyenne est : ")

TP6 | Résolution approchée d’équations

Les algorithmes classiques que nous verrons ici sont :

6.1 Introduction 60
6.2 Méthode de dichotomie 61
6.3 Approximation par une suite itérative 62
6.4 Exercices 64
6.5 Complément : Méthode de Newton 65
6.6 Complément : Méthode de la sécante 66
6.7 Quelques corrigés 68

Avertissement : Ce TP
comporte un peu plus
d’éléments théoriques
que les autres, ceux-
ci pourront être lus à
tête reposée ultérieu-
rement. Les exercices
comportent certaines
questions purement
théoriques qui ne né-
cessitent donc aucun
code.

6.1 Introduction
La résolution d’équation est un thème central des mathématiques. Si l’on connait
la résolution des équations polynomiale du premier, ou deuxième degré (pour

rappel : en Scilab , tout
est approché, mais le
même algorithme dans
un langage de calcul
formel donnerait une
solution exacte.

lesquelles nous avons déjà écrit des algorithmes de résolution exacte).

Le programme d’ECS1 mentionne le calcul approché de la racine d’une équation du
type f(x) = 0 dans la liste de savoir-faire exigibles en première année. On utilisera
différentes méthodes dont certaines résulteront d’une étude mathématique (suites
récurrentes, encadrements, dichotomie).

Nous allons voir dans ce TP différents algorithmes de résolution approchée de
l’équation f(x) = 0, où f est une fonction continue sur un intervalle [a; b] et pour
laquelle on sait que f(a) et f(b) sont de signe contraires.

Si au concours la partie algorithmique portait sur le thème de la résolution d’équa-
tion, celle-ci s’insérerait dans un problème plus général sur les suites et la ré-
solution d’équation, nous en profiterons donc pour rappeler quelques résultats
théoriques, que vous aurez peut-être besoin de replacer dans les questions précé-
dentes.

Exercice 47
1. Pourquoi f(x) = 0 et pas les autres équations ?
2. Pourquoi les hypothèses ci-dessus permettent-elles de conclure l’existence

d’une solution α à l’équation f(x) = 0. Énoncez le théorème.
3. Quelle hypothèse supplémentaire permettrait d’assurer l’unicité de la solu-

tion ?
4. Quel test (calcul) permet de vérifier que f(a) et f(b) sont de signe contraires.

6.2 Méthode de dichotomie
Normalement déjà abordée au lycée, c’est la méthode classique.

Principe

Initialisation : Connaissant deux valeurs a0 et b0 dont les images ont des signes
contraires, on sait donc que f s’annule entre ces deux valeurs en un réel x0.

Itération : Supposons connus an et bn tels que f(an)f(bn) < 0, c’est à dire
encadrant la solution cherchée. On calcule alors cn = an+bn

2 et on en calcule
l’image.
Si f(an) et f(cn) ont même signe, c’est que la solution se trouve entre cn et bn.
On pose alors an+1 = cn et bn+1 = bn.
Dans le cas contraire, la solution se trouve entre an et cn. On pose alors an+1 =
an et bn+1 = cn.

Arrêt : On s’arrête quand on a atteint la précision souhaitée, c’est à dire quand
bn − an est inférieure à celle-ci.

Justification de la convergence

Par construction, pour tout entiern, an ≤ bn. De plus, la suite (an) est croissante
et la suite (bn) est décroissante. En effet, Pour tout n, on a :

an ≤
an + bn

2 ≤ bn

Ainsi, que l’on ait an+1 = an ou an+1 = an+bn
2 , on a bien an+1 ≥ an.

De même, on montre que bn+1 ≤ bn

On peut le prouver par
récurrence

D’autre part, lim
n→∞

(bn − an) = lim
n→∞

b0 − a0
2n = 0.

Les suites (an) et (bn) sont donc adjacentes, et par suite convergentes vers une
même limite l. Puisque pour tout n, on a an ≤ x0 ≤ bn, en passant à la limite,
on obtient l ≤ x0 ≤ l et donc l = x0

Mise en œuvre

On suppose implémentée une fonction TP simulant une fonction réelle f pour
laquelle on connaît deux valeurs a et b encadrant une solution de l’équation. Le
script ci correspond à l’implémentation SCILAB de l’algorithme de dichotomie.

Cet algorithme utilise
certaines valeurs plu-
sieurs fois, et effectue
donc des calculs re-
dondants. On pourrait
améliorer l’algorithme
en stockant les valeurs
dans deux autres va-
riables. Toutefois, cette
version me parait plus
lisible et convient pour
les besoins d’ECS1.

e = input("précision : ")
a = valeurinferieure
b = valeursuperieure
while b - a > e

c=(a + b)/2
if TP(a)*TP(c) > 0 then

a=c
else

b=c
end

end
disp(a,"la solution est environ ")

Exercice 48
⋆ Prenez quelques instants pour commenter le code ci-dessus avec // et vos

remarques pour vous souvenir de la démarche.

⋆ Quand vous avez l’algorithme en tête, cachez le (si si). Vous pouvez maintenant
écrire (vous avez caché l’algorithme ?) une fonction tp6 qui calcule f(x) =
x3 + 2x2 + 3x + 4 et écrivez (toujours sans regarder !) un algorithme de
dichotomie pour approcher la racine de cette fonction en prenant comme
valeurs initiales a = −2 et b = −1.

dans un sujet, on fait
ceci avant d’écrire un
algorithme

⋆ Si vous avez terminé en avance…
Démontrez que l’équation f(x) = 0 admet une unique solution dans [−2;−1]

6.3 Approximation par une suite itérative
Un grand nombre de problèmes sur les suites consiste en la donnée d’une fonction

⋆ Pourquoi les mots
en gras sont-ils
importants ?

φ continue et d’un réel u0. On étudie la suite définie par :

{
u0 ∈ ℝ
un+1 = f(un)

On sait alors que si (un) converge vers une limite l, celle-ci vérifie φ(l) = l.

Exercice 49 On veut déterminer une valeur approchée de √2. Pour cela, on
cherche à résoudre l’équation (ℰ) 𝓍2 − 2 = 0 avec une suite itérative.
1. Montrer que (ℰ) équivaut à φ(x) = x avec φ : x ↦ 2x − 2x
2. On pose u0 := 2 et un+1 = φ(un). Taper la fonction récursive suite sui-

vante qui prend en entrée un nombre n et calcule un. Faire quelques essais et
constater que la suite ne converge pas.

function y=phi(x)
y=2*x-2/x

endfunction

function u=suite(n)
if n==0 then

u=2
else

u=phi(suite(n-1))
end

endfunction

Il serait plus efficace de
calculer les valeurs au
fur et à mesure avec la
fonction phi. Toutefois,
le code est plus simple
avec cette fonction
suite. Faisons donc
comme si les calculs
étaient instantanés.

3. Modifier la fonction phi pour qu’elle renvoie φ(x) = 1
3 (2x +

2
x). Vérifiez

rapidement que √2 est bien un point fixe de φpuis écrivez un programme
approx qui prend en entrée une précision p et renvoie une approximation de a
à la précision p ainsi que le nombre d’itérations nécessaires pour arriver à cette
précision. On pourra utiliser la fonction suitephi définie précédemment.

4. Même question avec φ(x) = 1
2 (x +

2
x)

6.4 Exercices

Exercice 50 : Une résolution par dichotomie
1. Démontrer que l’équation tanx − x = 0 admet une unique solution α ∈ ]

π
2 ;

3π
2 [.

2. Vérifier que 4 < α < 4, 7 puis écrire une procédure Scilab qui détermine, par dichotomie, une valeur
approchée à 10−4 près de α.

On pourra écrire dans un premier temps une fonction qui calcule tanx − x pour une entrée x.

Exercice 51 Une résolution par itération
1. Démontrer que l’équation cos x − x = 0 admet une unique solution α ∈ [0; 1].

2. On définit la suite (un) par u0 = 0 et un+1 = cos(un)

a. Prouver que ∀n ∈ ℕ, |un − α| ≤ (sin(1))n.

b. Déduire la convergence de (un) vers α

3. Écrire un programme permettant de calculer une valeur approchée de α à 10−6

Exercice 52 : Une équation, deux méthodes

On s’intéresse ici à l’équation f(x) − x = 0 où f est la fonction définie sur ℝ par
∀x ∈ ℝ, f(x) = (x2 + 1) e−x

1. Démontrer que l’équation f(x) − x = 0 admet une unique solution α ∈ ]
1
2 ; 1[.

On rappelle que 2 < e < 3

2. Écrire un programme qui calcule par dichotomie une valeur approchée de α à 10−6 près.

3. Nous allons maintenant approcher α par itérations successives.
On considère la suite définie par : u0 = 1 et ∀n ∈ ℕ, un+1 = f(un)

a. Prouver que ∀x ∈ [
1
2; 1] , |f′(x)| ≤

1
4 et 1

2 ≤ f(x) ≤ 1

b. Montrer que ∀n ∈ ℕ, |un − α| ≤ (
1
4)

n

c. En déduire que (un) converge vers α

d. Écrire un programme utilisant la suite (un) pour approcher α à 10−6 près.

6.5 Complément : Méthode de Newton

Figure 6.1: Méthode de
Newton pour résoudre
les équations.

Cette méthode n’est pas explicitement au programme. Néanmoins, tous les élé-
ments nécessaires à son étude sont au programme de première année. Nous allons
donc en voir les grandes lignes ainsi que l’algorithme d’application.

L’algorithme se comprend bien graphiquement. Partant d’un point pas trop éloigné
de la racine, on suit la tangente jusqu’à l’axe des abscisses, ce qui nous rapproche
du point d’intersection de la courbe de f avec (Ox). On remonte jusqu’à la courbe
et on suit à nouveau la tangente…

Exercice 53 Supposons connue une valeur approchée xn de la solution. Détermi-
ner l’abscisse xn+1 du point d’intersection de la tangente à 𝒞𝒻 au point d’abscisse
xn avec l’axe des abscisses.

Méthode de Newton :
On suppose f de classe 𝒞2, convexe et croissante (f′ > 0, f″ > 0) sur un
intervalle [a; b] dans lequel f s’annule en r (unique d’après les hypothèses).
Alors la suite définie par x0 = b et xn+1 = xn − f(xn)

f′(xn)
est strictement

décroissante et de limite r.

Démonstration : Posons φ : x ↦ x − f(x)
f′(x) de sorte que xn+1 = φ(xn). On a

φ′ = ff″
f′2 > 0 d’après les hypothèses donc φ est strictement croissante et la suite

6.1 récurrence classique
qu’il faut savoir rédiger

(xn) est monotone. Puisque x1 = f(x0) = f(b) = b− f(b)
f′(b) < b = x0, et donc

1 pour
tout entier n, r < xn+1 < xn. La suite est décroissante, minorée donc convergente
vers l’unique point fixe r.
La formule de Taylor-Lagrange appliquée à f en xn assure l’existence de ξ tel que :

∀n ∈ ℕ, f(xn + (r − xn)) = f(xn) + (r − xn)f′(r) + (r − un)2
f″(ξ)
2

(xn − r)f′(xn) − f(xn) = (r − un)2
f″(ξ)
2

xn − r −
f(xn)
f′(xn)

= (r − un)2
2f″(ξ)
f′(xn)

|xn+1 − r| ≤ |r − un|2
|f″(ξ)|
2|f′(xn)|

donc ∀n ∈ ℕ, |xn − r| ≤
1
K(K|x0 − r|)

2n

On a posé ici K =
max
[a;b]

f″

2min
[a;b]

f′ qui est bien défini car f′ et f″ sont continues donc bornées

sur le segment [a; b].

Bref, la convergence est très rapide (quadratique) si x0 − r est suffisamment petit.

Exercice 54 : Programmer la méthode de Newton pour résoudre l’équation x +
ex − 2 = 0 en prenant comme point de départ x0 = 1.

Et si vous êtes en avance, prenez donc quelques instants pour vérifier que la fonction
vérifie bien les hypothèses.

Voici un code dont vous pouvez vous inspirer. Il est agrémenté de quelques com-

Pour construire le gra-
phique, on a besoin de
la fonction f, appelée ici
fonction et de sa déri-
vée f′ appelée derivee.
Mais s’il s’agit unique-
ment d’approcher la
solution de l’équation,
on pourra préférer une
fonction phi qui code
f
f′

mandes graphiques et commence à 10 pour voir plus de choses.

function y=fonction(x)
y=(x+exp(x)-2)

endfunction

function y=derivee(x)
y=(1+exp(x))
endfunction

clf()
x=[0.5:0.05:10];plot(x,fonction(x))
x=10
a=[x;x]
b=[0;fonction(x)]
plot2d(a,b)
for i=1:10

x=x-fonction(x)/derivee(x)
a=[a;x;x];b=[b;0;fonction(x)]

end
plot2d(a,b)

Figure 6.2: Méthode de
la sécante.

6.6 Complément : Méthode de la sécante
Également appelée « Méthode de Lagrange », nous allons d’abord en illustrer le
principe avant de voir dans quelles conditions elle est applicable.

Principe

Le principe se comprend très bien visuellement. Connaissant un encadrement
[a; b] de la solution r cherchée, on trace la corde reliant A(a; f(a) à (b; f(b)) et
l’abscisse x1 du point d’intersection de la corde avec l’axe des abscisses est plus
proche de r que a. On itère le procédé.

Méthode de la sécante :
On considère un intervalle [a, b] et une fonction f de classe 𝒞2 sur [a, b]
strictement convexe (f″ > 0) vérifiant f(b) > 0 et f(a) < 0.
La suite (xn) définie par x0 := a et xn+1 := xnf(b) − bf(xn)

f(b) − f(xn)
converge

en croissant vers r.

Démonstration : On a a < r < b. Supposons construit xn tel que a < xn < r < b.
Alors xn+1 = λxn + μb avec λ = f(b)

f(b)−f(xn) > 0 et μ = −f(xn)
f(b)−f(xn) > 0 qui

vérifient λ + μ = 1.
Ainsi, xn+1 ∈ ]xn; b[ et, par convexité de f,
0 = λf(xn) + μf(b) ≥ f(λxn + μb) = f(xn+1).
Puisque r est l’unique valeur pour laquelle f s’annule,
on en déduit xn ≤ xn+1 ≤ r ≤ b. La suite est donc croissante et majorée par r donc
convergente vers un réel l ≤ r qui vérifie : l = lf(b) − bf(l)

f(b) − f(l) qui donne f(l) = 0
puisque l < b puis l = r

Voici un exemple de la méthode de la sécante appliquée à la fonction
x ↦ x2 + ex − 3 avec a = 0 et b = 3

function y=fonction(x)
y=(x^2+exp(x)-3)

endfunction

clf()
a=0
b=3
x=[0:0.05:b];plot(x,fonction(x))
x=a
M=[a;b]
N=[fonction(a);fonction(b)]
plot2d(M,N)
for i=1:10

x=(x*fonction(b)-b*fonction(x))/(fonction(b)-fonction(x))
M=[x;b]
N=[fonction(x);fonction(b)]
plot2d(M,N)

end

Exercice 55 Adapter le code ci-dessus pour qu’il donne une solution approchée
de l’équation x2 − x − 6 = 0 avec a = −3 et b = 0 puis avec a = 0 et b = 5. Le
premier cas ne correspond pas tout à fait aux hypothèses présentées ci-avant mais
l’algorithme converge tout de même.

6.7 Quelques corrigés

Solution pour les exercices du début de la séquence

Exercice 2 :
Approche de la solution de l’équation f(x) = 0 comprise dans l’intervalle [−2;−1],
où f(x) = x3 + 2x2 + 3x + 4.

function y=TP(x)
y=x^3+2*x^2+3*x+4

endfunction

e = input("précision : ")
a = -2
b = -1
while b - a > e

c=(a + b)/2
if TP(a)*TP(c) > 0 then

a=c
else

b=c
end

end
disp(a,"la solution est environ ")

Exercice 4 :

Quelle que soit la fonc-
tion qu’on utilise pour
définir la suite, le code
est le même pour calcu-
ler un

Voici les trois fonctions dont on a besoin pour tester, légèrement différentes.
function y=phi(x)

y=2*x-2/x
endfunction

function y=phi(x)
y=(2*x-2/x)/3

endfunction

function y=phi(x)
y=(2*x-2/x)/2

endfunction

function u=suite(n)
if n==0 then

u=2
else

u=phi(suite(n-1))
end

endfunction

Quand la suite est convergente, on obtient le nombre d’itérations nécessaires pour
approcher √2 à moins de 10−6 avec :

n=0
while suite(n)-sqrt(2)>=10^(-6)

n=n+1
end
disp(n)

Même code pour les deux mais on trouve



TP6 | Résolution approchée d’équations

69 Jean-Marc BERROU - 2014-2015

Corrigé succinct des exercices de la section 4

Je livre ici une proposition de code pour la partie algorithmique. Si j’ai le temps,
je compléterai la partie mathématique ultérieurement.

La seule chose qui
change, c’est la fonc-
tion, tout le reste de
l’algorithme est iden-
tique (à part les valeurs
initiales de a et b, et la
précision.

Dichotomie sur tanx − x = 0

function y=f(x)
y=tan(x)-x

endfunction

a=4
b=4.7
while b-a>=10^(-4)

c=(a+b)/2
if f(a)*f(c)>0 then

a=c
else

b=c
end

end

disp(a) //ne pas oublier

Itération sur cos x − x = 0

u=0
n=0

while sin(1)^n>=10^(-6)
n=n+1
u=cos(u)

end

disp(u)

Ici encore, l’algorithme n’est pas du tout op-
timal, on pourrait stocker les valeurs succes-
sives de sin(1)n dans une variable pour
réduire le nombre de calculs.

Une équation, deux méthodes

Il est souvent plus pra-
tique de définir une
fonction avant l’algo-
rithme. Attention, la
fonction ne sera pas la
même si l’algorithme
utilise la dichotomie ou
une itération vers un
point fixe.

Dichotomie :

function y=f(x)
y=(x^2+1)*exp(-x)-x

endfunction

a=0.5
b=1
while b-a>=10^(-6)

c=(a+b)/2
if f(a)*f(c)>0 then

a=c
else

b=c
end

end

disp(a)

Itérative :

function y=f(x)
y=(x^2+1)*exp(-x)

endfunction

u=0
n=0
while 0.25^n>=10^(-6)

n=n+1
u=f(u)

end

disp(u)

TP7 | Variables aléatoires

Les probabilités sont un des impondérables du programme de mathématiques, qui
se prête particulièrement bien à l’expérimentation en général, informatique en
particulier.

7.1 Simulation de variables aléatoires discrètes 70
7.2 Exercices 72
7.3 Simulation de variables aléatoires continues 75
7.4 Phénomènes limites. 76
7.5 Échantillonnage 76
7.6 Solution des exercices 77

7.1 Simulation de variables aléatoires discrètes
Nous avons déjà vu la commande rand et utiliserons plus ici la commande grand,
qu’il faut entendre comme g-rand, generate randomness, qui génère des nombres
aléatoires de meilleure manière, et surtout avec plus d’options que son analogue

dans tous les algo-
rithmes qui suivent,
les commandes sont
compatibles avec les
vecteurs ».

Ainsi, rand() peut
être remplacé par
rand(5,7) par exemple
si l’on souhaite une
matrice de valeurs aléa-
toires.

rand.

grand permet de simuler directement les principales lois de probabilités discrètes
ou continues, et de les simuler en grand nombre sous forme de matrices pour des
simulations, estimations… C’est la commande idéale, prête à l’emploi.

Toutefois, il peut être intéressant de savoir les implémenter à partir d’une loi uni-
forme sur [0; 1] obtenue avec la commande rand(). Dans les exemples ci-dessous,
le premier code est la simulation avec rand(), le second sa contrepartie grand.

Loi de Bernoulli

Pour simuler une expérience de Bernoulli qui renvoie 1 avec une probabilité
p ∈]0; 1[ contenue dans une variable p avec la commande :

X=(rand()<p)

X contient T avec une probabilité p ou F qui peuvent respectivement être utilisés
comme 1 et 0 dans d’éventuels calculs.

Loi uniforme sur [[1 ;n]]

Si n contient la valeur d’un nombre entier, le code suivant…

X=floor(n*rand())+1
Y=grand(m, n, "uin", min, max)

X contient un nombre entier au hasard entre 1 et n inclus.
Y est une matricem×n de nombres aléatoires suivant 𝒰 ([[min;max]])

Loi binomiale

X=sum(rand(1,n)<p)
Y=grand(m, n, "bin", N, p)

X compte le nombre de succès obtenus lors de n répétitions d’une épreuve de
Bernoulli de paramètre p.
Y est une matricem×n de nombres aléatoires suivant ℬ(n, p)

Loi géométrique

while rand()>=p
X=X+1

end

X compte le nombre de tirages nécessaires suivant une loi de Bernoulli de para-
mètre p avant obtention du premier succès.

Y = grand(m, n, "geom", p)

Y est une matrice m × n de nombres aléatoires suivant la loi géométrique de
paramètre p

Loi de Poisson

Y = grand(m, n, "poi", mu)

Y est une matricem×n de nombres aléatoires suivant

7.2 Exercices

Exercice 56
1. Une urne contient B boules blanches etN boules noires (N ≤ B). Compléter la

fonction ci-dessous qui prend en entrée trois nombres : n ≤ N,N et B, simule
le tirage sans remise de n boules dans l’urne puis renvoie le nombre de boules
noires obtenues.

function c=noires(n,N,B)
c=0
for i=1:n

if (N+B)*rand()<=N then
_ _ _ _ _ _ _ _ _ _
_ _ _ _ _ _ _ _ _ _

else
_ _ _ _ _ _ _ _ _ _

end
end
endfunction

2. Que fait le programme ci dessous ?

n=10
N=10
B=20
clf()
S=zeros(1,N+1)
for i=1:10000

r=noires(n,N,B)+1
S(r)=S(r)+1

end
S=S/10000
bar(0:N,S)

3. Si vous avez de l’avance… le nombre n de boules tirées a été restreint à n ≤ N
pour éviter la possibilité d’épuiser les boules noires. Modifier la fonction pour
qu’elle prenne en compte l’éventualité d’arriver au bout d’une des deux couleurs
de boules. Ainsi, on peut choisir n ≤ N+ B

Exercice 57 On considère une suite de lancers successifs et indépendants d’une
pièce de monnaie. On suppose que la probabilité p d’obtenir Pile est de 2

3 . On
note X le rang d’apparition du premier « double pile » (c.a.d. le rang d’apprition
du deuxième des deux piles)
Compléter le code suivant pour qu’il demande un entier n et simule n fois l’expé-
rience ci-dessus puis calcule le nombre moyen de tirages nécessaires à l’obtention
de la première paire de Pile.
On code pile par 1 (ou %T) et face par 0 (ou %F)

n = input('Nombre de simulations : ')
S = zeros(1,n)
for i = 1:n

k = 0
x = 0
y = 0
while x*y == 0

_ _ _ _ _ _ _ _ _ _ _ _ _ _
_ _ _ _ _ _ _ _ _ _ _ _ _ _
_ _ _ _ _ _ _ _ _ _ _ _ _ _

end
S(i) = k

end
moyenne = _ _ _ _ _ _ _ _ _ _ _ _ _ _
disp (moyenne, 'le nombre moyen de tirage est :')

Exercice 58 : ESC1999
Soit n ∈ ℕ⋆. Une urne Un contient n boules numérotées de 1 à n. On effectue
dans cette urne une succession de tirages d’une boule, en appliquant la règle
suivante : si une boule tirée porte le numéro k, avant de procéder au tirage suivant,
on enlève toutes les boules dont le numéro est supérieur ou égal à k.
On note Xn la variable aléatoire égale au nombre de tirages nécessaires pour vider
l’urne Un de toutes ses boules.
Compléter le programme suivant afin qu’il simule l’expérience :

// d'après ESC1999
b = input('Entrez le nombre initial de boules ')
tirages=0
while _ _ _ _ _

numero= _ _ _ _ _ _ rand() _ _ _
b= _ _ _
tirages = _ _ _ _ _ _

end
disp ('tirages' , tirages , 'L''urne est vidée en ')

Exercice 59 : ÉCRICOME ECE 2015
L’urne U1 contient N boules : (N − 1) boules blanches et 1 boule noire.
On effectue des tirages sans remise dans l’urne U1, jusqu’à l’obtention de la
boule noire. On note X la variable aléatoire qui prend pour valeur le nombre de
tirages nécessaires pour l’obtention de la boule noire.
1. On simule 10000 fois cette expérience aléatoire.

Recopier et compléter le programme Scilab suivant pour qu’il affiche l’histo-
gramme donnant la fréquence d’apparition du rang d’obtention de la boule
noire :

N = input('Donner un entier naturel non nul');
S = zeros(1,N)
for k = 1:10000

i = 1;
M = N;
while _ _ _ _ _ _ _ _ _ _ _ _ _ _

i = i+1
M = _ _ _ _ _ _ _ _ _ _ _ _ _ _

end
S(i) = S(i) + 1;

end
disp (S/10000)
bar (S/10000)

2. On exécute le programme complété ci-dessus. On entre 5 au clavier et on
obtient l’histogramme suivant :

Quelle conjecture pouvez-vous émettre sur la loi de la variable aléatoire X ?

Exercice 60 : Paradoxe du Duc de Toscane
On lance trois dés simultanément et on appelle X la variable aléatoire égale à la
somme des trois numéros obtenus.
1. Écrire une fonction qui simule X.

2. Compléter cette fonction en un programme qui effectue 20000 lancers et
compte le nombre fois où X vaut 9 et celui où X vaut 10.

Remarque : Le paradoxe vient du fait qu’il y a exactement le même nombre de
décompositions pour obtenir 9 ou 10 (il y en a 6) mais que ℙ(X = 9) ≠ ℙ(X = 10).
Saurez-vous résoudre ce paradoxe ?

Exercice 61 : Petit défi bonus, redresser une pièce truquée
Supposons que vous disposez d’une fonction cheat qui simule une expérience de
Bernoulli de paramètre 1

3 (une pièce truquée). Écrivez un code qui simule le lancer
d’une pièce bien équilibrée à partir de cette « pièce truquée ».
Indication : Si vous lancez plusieurs fois la pièce, certains événements sont équipro-
bables… Indication

7.3 Simulation de variables aléatoires continues
Pour simuler une loi de probabilité continue dont la densité est non non nulle
sur un intervalle ouvert I et nulle en dehors, utiliser sa fonction de répartition,

7.1 à vérifier ! qui sera continue et strictement croissante sur I1 et réalisera donc une bijection
de I sur ]0; 1[. Dans le cas de la loi exponentielle, nous pouvons expliciter une
réciproque.

Loi uniforme continue

Y = grand(m, n, "unf", Low, High)

Y est une matricem×n de nombres aléatoires suivant

Loi exponentielle

Pour simuler une loi de probabilité continue dont la densité est non non nulle
sur un intervalle ouvert I et nulle en dehors, utiliser sa fonction de répartition,

7.2 à vérifier !qui sera continue et strictement croissante sur I2 et réalisera donc une bijection
de I sur ]0; 1[. Dans le cas de la loi exponentielle, nous pouvons expliciter une
réciproque.

Y = grand(m, n, "exp", E)

Y = grand(m, n, "exp", E)

Y est une matrice m × n de nombres aléatoires suivant une loi exponentielle
d’espérance E = 1

λ

Loi normale

Y = grand(m, n, "nor", mu, sigma)

Y est unematricem×n de nombres aléatoires suivant une loi normale de moyenne
mu et d’écart-type sigma

7.4 Phénomènes limites.

Simulation de ℬ(n; λn) et comparaison avec la loi de Poisson

7.5 Échantillonnage

7.6 Solution des exercices

Exercices sur les lois discrètes

tirage sans remise de n boules parmi N noires et B blanches

function c = noires(n,N,B)
c = 0
for i = 1:n

if (N+B)*rand()<=N then
N = N-1
c = c+1

else
B = B-1

end
end
endfunction

Le deuxième code simule 10000 expériences avec 10 boules noires, 20 boules
blanches, parmi lesquelles on tire 10 boules, remplit au fur et à mesure un vec-
teur qui contient le nombre d’apparition de chaque valeur possible, puis calcule
la fréquence d’apparition de chaque valeur et affiche le résultat sous forme de
diagramme en barres.

lancers jusqu’à Pile-Pile

n = input('Nombre de simulations : ')
S = zeros(1,n)
for i = 1:n

k = 0
x = 0
y = 0
while x*y == 0

x = y
y = (3*rand()<2)
k = k+1

end
S(i) = k

end
moyenne = sum(S)/n
disp (moyenne, 'le nombre moyen de tirage est :')

ESC1999

// d'après ESC1999
b=input('Entrez le nombre initial de boules ')
tirages=0
while b>0

numero = floor(b*rand()) + 1

disp(numero)
b= min(numero-1,b)
tirages = tirages + 1

end
disp('tirages',tirages,'L''urne est vidée en ')

ÉCRICOME ECE 2015

N = input('Donner un entier naturel non nul');
S = zeros(1,N)
for k = 1:10000

i = 1;
M = N;
while M*rand()>1

i = i+1
M = M-1

end
S(i) = S(i)+1;

end
disp (S/10000)
bar (S/10000)

Il semble que X suive une loi uniforme sur [[1; 5]]



TP7 | Variables aléatoires

79 Jean-Marc BERROU - 2014-2015



ECS 1 - Externat des Enfants Nantais [2014-2015]

Jean-Marc BERROU - 2014-2015 80

TP8 | Intégration

Les intégrales sont partout dans les mathématiques appliquées. Certaines fonc-
tions, même très classiques comme x ↦ exp(−x2) n’ont pourtant pas de primitive
explicite. Dans ce cas, ou dans le cas de fonctions définies par une table de valeur
relevées expérimentalement par exemple, les méthodes numériques sont néces-
saires pour calculer l’intégrale sur un intervalle de ces fonctions.

1 cherchez par exemple
Calcul numérique d’une
intégrale sur Wikipedia
pour vous faire une idée

Il existe un grand nombre1 de méthodes pour approcher ∫
b
a f(x)dx. Elles diffèrent

par leur complexité, leur rapidité de convergence, leur précision…

Nous verrons ici :

8.1 Méthode des Rectangles 80
8.2 Méthode de Monte Carlo 83
8.3 Propositions de solutions 85

8.1 Méthode des Rectangles

Figure 8.1: rectangles
à gauche

Le principe de l’algorithme est simple :

⋆ on subdivise l’intervalle [a; b] en x0 = a, x1, … , xn−1, xn = b, par
exemple avec la commande linspace(a,b,n+1). Nous avons donc
n + 1 valeurs qui définissent n intervalles.

⋆ on calcule les hauteurs des rectangles (images par f des xi)
⋆ on somme les aires des rectangles.

Figure 8.2: rectangles
à droite

Il y a trois manières de définir les rectangles :
1. méthode dite « rectangles à gauche »
2. méthode dite « rectangles à droite »
3. méthode dite « du point médian » qui est la plus efficace.
Un petit dessin valant mieux qu’un long discours, en voici l’illustration ci-contre
où on a appliqué les méthodes de rectangles à l’application x ↦ x2 − 1 :

Figure 8.3: méthode du
point milieu

les formules

Les formules (qui ne seront pas démontrées ici) utilisées dans l’algorithme sont :



b

a
f(x)dx = lim

n→∞
b − a
n

n−1


k=0

f(a + kb − an )

= lim
n→∞

b − a
n

n


k=1

f(a + kb − an )

= lim
n→∞

b − a
n

n−1


k=0

f
(
a + (k + 12)

b − a
n )

La méthode du point
milieu est visuellement
plus équilibrée et on
peut démontrer mathé-
matiquement qu’elle
est plus efficace que les
deux autres.

Exercice 62 Saurez-vous retrouver quelle formule correspond à quelle méthode
(gauche, droite, milieu ?)

Disons que l’on souhaite calculer ∫
2
0 e

x2dx.

8.2 attention à bien
appliquer les points,
comme .^, ou ./, etc.

On commence par définir la fonction f, d’une manière compatible avec les vecteurs2,
puis une subdivision de l’intervalle d’intégration avec x=linspace(0,2,101).
Nous avons donc ici n = 100 intervalles et un vecteur x comportant n+1 valeurs.
Onmodifie alors x pour ne garder que les valeurs utiles à la définition des « rectangles ».
⋆ x=x(1:100) pour les rectangles à gauche

⋆ x=x(2:101) pour les rectangles à droite

8.3 vous êtes bien
convaincu que ça
convient ?

⋆ x=0.5*(x(1:100)+x(2:101)) pour le point milieu3
on peut alors calculer y=f(x) la valeur approchée de l’intégrale est donnée par
2*sum(y)/100

Quand vous expérimentez sur l’intégration avec Scilab , vous pouvez utiliser la
commande integrate pour vérifier la précision de vos calculs. Celle-ci n’est pas
au programme mais fait appel à des algorithmes optimisés pour calculer votre
intégrale et peut donc être considérée comme une référence fiable.

X=integrate('f(x)','x',0,2);X

Voici un code facilement adaptable à tous les cas :

function y=f(x)
y=exp(x.^2)

endfunction

n=100000;
a=0;
b=2;
x=linspace(a,b,n+1);
x=x(1:n); // ou x=x(2:n+1)
// ou x=0.5*(x(1:n)+x(2:n+1))
y=f(x);
integrale=(b-a)/n*sum(y)

Pour cette fonction, positive et croissante, votre résultat sera systématiquement
inférieur avec les rectangles à gauche, supérieur avec les rectangles à droite, et
mieux équilibrée avec la méthode du point milieu. Dans tous les cas, l’approxi-
mation sera d’autant meilleure que n sera grand. Cela se comprend bien sur un
dessin.

N’hésitez pas à tester avec des valeurs importantes comme 1000, 10000 voire
100000, le calcul reste rapide.

Résumé des étapes

Pour mettre en œuvre la méthode des rectangles, on pourra suivre les étapes
suivantes :
1. définir la fonction à intégrer (de manière vecteur-proof)
2. définir la subdivision x de l’intervalle d’intégration en n valeurs

x=linspace(a,b,n+1)
3. modifier x en vecteur des valeurs à utiliser selon le type de rectangle choisi

x=x(1:n) pour les rectangles à gauche
4. calculer le vecteur y des valeurs correspondantes y=f(x)
5. appliquer la formule de Riemann integrale=sum(y)*(b-a)/n

Exercice 63
1. Écrire une fonction Scilab qui modélise la densité de la loi normale centrée

réduite f : x ↦ 1
√2π e

−x2
2

2. On rappelle que ∫
0
−∞

1
√2π e

−x2
2 dx = 1

2 , écrire une fonction Scilab qui modé-

lise la fonction de répartition F : x ↦ ∫
x
−∞

1
√2π e

−t2
2 dt

3. Comparer avec ce que l’on peut obtenir en utilisant la commande integrate

Exercice 64 Écrire un programme Scilab qui demande un nombre p en entrée,
et calcule pour toutes les valeurs de k ∈ [[1; p]] le nombre d’itérations nécessaires
pour arriver à une approximation de ∫

1
0 3x

2dx avec une précision 10−k.
On rappelle que cette intégrale vaut 1

Remarque sur la vitesse de convergence

On peut démontrer que les méthodes des rectangles à gauche et à droite ont une
précision en C

n où C est une constante dépendant de la longueur de l’intervalle
d’intégration et de la fonction.
La méthode du point milieu offre quant à elle une précision en C′

n2 , ce qui est
considérablement plus efficace.

8.2 Méthode de Monte Carlo
Monte Carlo, c’est un rocher, des princes et des princesses, un Grand Prix, un
Masters, RMC, TMC et … une méthode d’intégration numérique !

Le principe est le suivant : Disons que vous souhaitez déterminer la surface d’une
mare et que pour cela vous disposez de moyens illimités (d’une imagination illi-
mitée) qui vous permettent via un hélicoptère de disperser uniformément des
cailloux blancs dans un rectangle de surface connue S englobant la mare.

Quel est le rapport ?

Et bien justement, le rapport entre le nombre de cailloux tombés dans la mare,
appliqué à la surface S donne une bonne approximation de la surface de la mare
si le nombre de pierres lancées est suffisant.

On peut utiliser la même idée pour approcher la valeur d’une intégrale de fonction.

− On déterminen valeurs (xk; yx)1≤k≤n, les xk suivant une loi uniforme
sur [a; b] et les yk une loi uniforme sur f([a; b])

− Pour chaque couple de valeur (xk; yk), on compare yk et f(xk)

− on détermine la proportion de points sous la courbe que l’on multiplie
par l’aire du rectangle.

Le schéma ci-dessous illustre la méthode :

Exercice 65

On rappelle que l’équation d’un cercle de centre O et de rayon 1 est x2 + y2 = 1.

1. Écrire un programme qui utilise une méthode de typeMonte Carlo pour évaluer
la surface de ce disque (π).

2. Modifier le code pour déterminer le nombre d’itérations nécessaires pour
approcher πà moins de 10−4 près

Exercice 66

On rappelle que l’équation d’un cercle de centre O et de rayon 1 est x2 + y2 = 1.

1. Écrire un programme qui utilise une méthode de typeMonte Carlo pour évaluer
la surface de ce disque (π). On pourra prendre 1000 couples de valeurs

2. Modifier le code pour déterminer le nombre d’itérations nécessaires pour
approcher π à moins de 10−4 près

3. Modifier encore le code pour qu’il demande en entrée le nombre n de valeurs
dans la Méthode, puis évalue sur 1000 simulations distinctes l’écart-type par
rapport à π.

4. En remarquant que la courbe d’équation y = √1 − x2 est celle d’un demi
cercle de centre O et de rayon 1, comparer le nombre d’itérations nécessaires
pour obtenir une approximation de πde précision comparable avec une mé-
thode de rectangles sur point milieu ou méthode de Monte Carlo.

Remarque : La Méthode de Monte Carlo n’est pas particulièrement efficace pour
calculer les intégrales auxquelles on peut se retrouver confronté en prépa. Elles sont
toutefois concrètement utilisées pour des problèmes de dimensions plus grande, avec
par exemple de nombreux paramètres comme pour des décisions stratégiques, ou en
finance

8.3 Propositions de solutions

Rectangles

Ici, on calcule l’intégrale entre 0 et x, que l’on ajoute ou soustrait à 1
2 en fonction

du signe de x.
En effet, on a :



x

−∞

1
√2π

e−
t2
2 dt = 

0

−∞

1
√2π

e−
t2
2 dt+

x

0

1
√2π

e−
t2
2 dt = 1

2+

x

0

1
√2π

e−
t2
2 dt

function y=f(x)
y=exp(-x.^2/2)/sqrt(2*%pi)

endfunction

function y=frepartition(x)
n=500
u=linspace(0,abs(x),n+1)
u=0.5*(u(1:n)+u(2:n+1))
valeurs=f(u)
integrale=sum(valeurs)*abs(x)/n
if x>=0 then

y=0.5+integrale
else

y=0.5-integrale
end
endfunction

La commande integrate ne donne rien de pertinent avec une borne inférieure
trop petite comme−1099. Vous pouvez vous convaincre que−100 suffit en tapant
X=integrate('f(x)','x',-100,1);X

Vous pourrez ensuite comparer par exemple le résultat des commandes :

X=integrate('f(x)','x',-100,a);X
frepartition(a)

pour a = 1, a = 4, a = −10, et toutes les valeurs que vous voulez.

Tant que l’on reste dans le domaine « utile » (à quelques σ de 0), les résultats sont
très précis avec une subdivision de taille 500. Au delà de 5σ, on gagnerait à optimiser
la subdivision en « resserrant » les valeurs autour de zéro. Ce type de découpage à
pas variable sort du cadre de ces TP.

Voici le code qui permet de représenter graphiquement le nombre d’itérations
nécessaires à l’obtention d’une précision donnée. Cette précision étant définie en
10k (1 ≤ k ≤ p), j’ai choisi de représenter log10(S).

function y=f(x)
y=3*x.^2

endfunction

clf()
p=input('précision maximale :')
S=zeros(1,p)
n=1
for k=1:p

integrale=0
while abs(integrale-1)>10^(-k)

n=n+1
x=linspace(0,1,n+1)
x=0.5*(x(1:n)+x(2:n+1))
integ=sum(f(x))/n

end
S(k)=n

end

bar(log10(S))

Noter la position du n=1. On ne remet pas le compteur à zéro à chaque fois, en
effet, on peut supposer que pour atteindre une plus grande précision, on aura
besoin d’un découpage plus fin. Pas besoin, donc, de refaire les premiers calculs.
Pour p = 7, on obtient le graphique suivant. Pour des valeurs plus grandes, le
programme tel quel peut prendre énormément de temps. On peut améliorer l’algo-
rithme en remplaçant n=n+1 par n=n+1000 si S(k-1) dépasse une certaine valeur.

8.4 la croissance linéaireSaurez vous interpréter ce graphe4 ? et à quoi sert le log10

Monte-Carlo

// approximation de pi
// par Montecarlo
// version vectorielle
n=20000
x=2*rand(1,n)-1
y=2*rand(1,n)-1
r=x.^2+y.^2 // distances à O
proportion=sum(r<1)/n
aire=proportion*4
disp(aire)

Si je trouve le code précédent plus élégant, on peut envisager une version itérative
dans l’optique de modifier le code pour la question 2.

// approximation de pi
// par Montecarlo
// version iterative
n=20000
compteur=0
for i=1:n

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
end
proportion=compteur/n
aire=proportion*4
disp(aire)

Pour la question 2…

// approximation de pi
// par Montecarlo
// nombre d'iterations avant approximation suffisante
precision=0.0001
compteur=0
nbiter=1
while abs(4*compteur/nbiter-%pi)>precision

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
nbiter=nbiter+1

end
disp(nbiter,'le nombre d''itérations nécessaires est : ')

Quelques tests vous montrerons que le résultat est très fluctuant. Notamment,
si on sait que la méthode converge, un bon résultat peut être momentanément
« empiré » par des tests supplémentaires.

Il est certainement plus sensé de procéder à de l’échantillonnage pour évaluer la
méthode, en s’arrêtant au bout d’un nombre arbitraire, mais relativement impor-
tant, de constater la précision atteinte, et de répéter le processus suffisamment
pour pouvoir effectuer des statistiques avec les données.

// évaluation de la précision de la Méthode de Monte Carlo
n=input('nombre d''itérations : ')
echantillon=[]
for i=1:100

compteur=0
for i=1:n

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
end
proportion=compteur/n
echantillon=[echantillon,proportion*4]

end
ecart=echantillon-%pi
ecarttype=sqrt(sum(ecart.^2))
disp(ecarttype)

Avec un échantillon de taille 100, j’ai obtenu un écart type de l’ordre de 0,12 pour
10000 itérations, et de l’ordre de 0,03 pour 100000 itérations (ne le lancez que si
vous aimez prendre un café devant votre ordinateur, c’est long…)

Moralité, on peut obtenir un bon résultat très rapidement, mais le nombre d’ité-
rations nécessaires pour obtenir un intervalle de fluctuation suffisamment petit
pour que le résultat soit exploitable est trop important. Aussi, on préférera une
méthode numérique de type « point milieu » ou« Simpson »(très couramment
utilisée).

TP9 | Intégration

Les intégrales sont partout dans les mathématiques appliquées. Certaines fonc-
tions, même très classiques comme x ↦ exp(−x2) n’ont pourtant pas de primitive
explicite. Dans ce cas, ou dans le cas de fonctions définies par une table de valeur
relevées expérimentalement par exemple, les méthodes numériques sont néces-
saires pour calculer l’intégrale sur un intervalle de ces fonctions.

1 cherchez par exemple
Calcul numérique d’une
intégrale sur Wikipedia
pour vous faire une idée

Il existe un grand nombre1 de méthodes pour approcher ∫
b
a f(x)dx. Elles diffèrent

par leur complexité, leur rapidité de convergence, leur précision…

Nous verrons ici :

1.1 But du cours et découverte du logiciel 4
1.2 Premières opérations 5
1.3 Logique et booléens 7
1.4 Chaînes de caractères 9
1.5 La vraie nature de Scilab 10
1.6 Un peu de dessin 11
2.1 Quelques rappels d’algorithmique 12
2.2 Bien communiquer, c’est important : entrées et sorties 14
2.3 Enregistrer son travail : Scinotes 14
2.4 Structures conditionnelles if ... then ... else 15
2.5 Quelques corrigés : 18
3.1 Boucle for ... to ... end 20
3.2 Boucle while ... condition 25
3.3 Exercices 26
3.4 Quelques corrigés : 28
4.1 Vecteurs 34
4.2 Matrices 38
4.3 Résolution de systèmes linéaires 41
4.4 Approfondissements 42
4.5 Un peu de culture G : Alan Turing et la pomme d’Apple 43
4.6 Quelques corrigés 46
5.1 Fonctions Scilab 48
5.2 Récursivité 50
5.3 Exercices 51
5.4 Culture G : L’autoréférence 54
5.5 Quelques corrigés : 56
6.1 Introduction 60
6.2 Méthode de dichotomie 61
6.3 Approximation par une suite itérative 62
6.4 Exercices 64
6.5 Complément : Méthode de Newton 65

6.6 Complément : Méthode de la sécante 66
6.7 Quelques corrigés 68
7.1 Simulation de variables aléatoires discrètes 70
7.2 Exercices 72
7.3 Simulation de variables aléatoires continues 75
7.4 Phénomènes limites. 76
7.5 Échantillonnage 76
7.6 Solution des exercices 77
8.1 Méthode des Rectangles 80
8.2 Méthode de Monte Carlo 83
8.3 Propositions de solutions 85
9.1 Méthode des Rectangles 91
9.2 Méthode de Monte Carlo 94
9.3 Propositions de solutions 96
10.1 Principales Commandes 100
10.2 Quelques codes à connaître 103
10.3 Convergences 106
10.4 Simulation de variables aléatoires classiques 107
10.5 Résolution d’équations, dichotomie 108
10.6 Exercices 109

9.1 Méthode des Rectangles

Figure 9.1: rectangles
à gauche

Le principe de l’algorithme est simple :

⋆ on subdivise l’intervalle [a; b] en x0 = a, x1, … , xn−1, xn = b, par
exemple avec la commande linspace(a,b,n+1). Nous avons donc
n + 1 valeurs qui définissent n intervalles.

⋆ on calcule les hauteurs des rectangles (images par f des xi)
⋆ on somme les aires des rectangles.

Figure 9.2: rectangles
à droite

Il y a trois manières de définir les rectangles :
1. méthode dite « rectangles à gauche »
2. méthode dite « rectangles à droite »
3. méthode dite « du point médian » qui est la plus efficace.
Un petit dessin valant mieux qu’un long discours, en voici l’illustration ci-contre
où on a appliqué les méthodes de rectangles à l’application x ↦ x2 − 1 :

Figure 9.3: méthode du
point milieu

les formules

Les formules (qui ne seront pas démontrées ici) utilisées dans l’algorithme sont :



b

a
f(x)dx = lim

n→∞
b − a
n

n−1


k=0

f(a + kb − an )

= lim
n→∞

b − a
n

n


k=1

f(a + kb − an )

= lim
n→∞

b − a
n

n−1


k=0

f
(
a + (k + 12)

b − a
n )



ECS 1 - Externat des Enfants Nantais [2014-2015]

Jean-Marc BERROU - 2014-2015 92

La méthode du point
milieu est visuellement
plus équilibrée et on
peut démontrer mathé-
matiquement qu’elle
est plus efficace que les
deux autres.

Exercice 67 Saurez-vous retrouver quelle formule correspond à quelle méthode
(gauche, droite, mileu ?)

Disons que l’on souhaite calculer
2
∫
0
ex2 .

On commence par définir la fonction f, d’une manière compatible avec les vecteurs,
puis une subdivision de l’intervalle d’intégration avec x=linspace(0,2,101).
Nous avons donc ici n = 100 intervalles et un vecteur x comportant n+1 valeurs.
Onmodifie alors x pour ne garder que les valeurs utiles à la définition des « rectangles ».
⋆ x=x(1:100) pour les rectangles à gauche

⋆ x=x(2:101) pour les rectangles à droite

9.2 vous êtes bien
convaincu que ça
convient ?

⋆ x=0.5*(x(1:100)+x(2:101)) pour le point milieu2
on peut alors calculer y=f(x) la valeur approchée de l’intégrale est donnée par
2*sum(y)/100

Quand vous expérimentez sur l’intégration avec Scilab , vous pouvez utiliser la
commande integrate pour vérifier vos calculs. Celle-ci n’est pas au programme
mais fait appel à des algorithmes optimisés pour calculer votre intégrale.
X=integrate('f(x)','x',0,2);X

Voici un code facilement adaptable à tous les cas :

n=100000;
a=0;
b=2;
x=linspace(a,b,n+1);
x=x(1:n); // ou x=x(2:n+1)
// ou x=0.5*(x(1:n)+x(2:n+1))
y=f(x);
integrale=(b-a)/n*sum(y)

Pour cette fonction, positive et croissante, votre résultat sera systématiquement
inférieur avec les rectangles à gauche, supérieur avec les rectangles à droite, et
mieux équilibrée avec la méthode du point milieu. Dans tous les cas, l’approxi-
mation sera d’autant meilleure que n sera grand. Cela se comprend bien sur un
dessin.

N’hésitez pas à tester avec des valeurs importantes comme 1000, 10000 voire
100000, le calcul reste rapide.

Exercice 68
1. Écrire une fonction Scilab qui modélise la densité de la loi normale centrée

réduite f : x ↦ 1
√2π e

−x2
2

2. On rappelle que ∫
0
−∞

1
√2π e

−x2
2 dx = 1

2 , écrire une fonction Scilab qui modé-

lise la fonction de répartition F : x ↦ ∫
x
−∞

1
√2π e

−t2
2 dt

3. Comparer avec ce que l’on peut obtenir en utilisant la commande integrate

Exercice 69 Écrire un programme Scilab qui demande un nombre p en entrée,
et calcule pour toutes les valeurs de k ∈ [[0; p]] le nombre d’itérations nécessaires
pour arriver à une approximation de ∫

1
0 3x

2dx avec une précision 10−k.
On rappelle que cette intégrale vaut 1

Remarque sur la vitesse de convergence

On peut démontrer que les méthodes des rectangles à gauche et à droite ont une
précision en C

n où C est une constante dépendant de la longueur de l’intervalle
d’intégration et de la fonction.
La méthode du point milieu offre quant à elle une précision en C′

n , ce qui est
considérablement plus efficace.

9.2 Méthode de Monte Carlo
Monte Carlo, c’est un rocher, des princes et des princesses, un Grand Prix, un
Masters, RMC, TMC et … une méthode d’intégration numérique !

Le principe est le suivant : Disons que vous souhaitez déterminer la surface d’une
mare et que pour cela vous disposez de moyens illimités (d’une imagination illi-
mitée) qui vous permettent via un hélicoptère de disperser uniformément des
cailloux blancs dans un rectangle de surface connue S englobant la mare.

Quel est le rapport ?

Et bien justement, le rapport entre le nombre de cailloux tombés dans la mare,
appliqué à la surface S donne une bonne approximation de la surface de la mare
si le nombre de pierres lancées est suffisant.

On peut utiliser la même idée pour approcher la valeur d’une intégrale de fonction.

− On déterminen valeurs (xk; yx)1≤k≤n, les xk suivant une loi uniforme
sur [a; b] et les yk une loi uniforme sur f([a; b])

− Pour chaque couple de valeur (xk; yk), on compare yk et f(xk)

− on détermine la proportion de points sous la courbe que l’on multiplie
par l’aire du rectangle.

Le schéma ci-dessous illustre la méthode :

Exercice 70
On rappelle que l’équation d’un cercle de centre O et de rayon 1 est x2 + y2 = 1.

1. Écrire un programme qui utilise une méthode de typeMonte Carlo pour évaluer
la surface de ce disque (π).

2. Modifier le code pour déterminer le nombre d’itérations nécessaires pour
approcher πà moins de 10−4 près

Exercice 71
On rappelle que l’équation d’un cercle de centre O et de rayon 1 est x2 + y2 = 1.
1. Écrire un programme qui utilise une méthode de typeMonte Carlo pour évaluer

la surface de ce disque (π). On pourra prendre 1000 couples de valeurs

2. Modifier le code pour déterminer le nombre d’itérations nécessaires pour
approcher πà moins de 10−4 près

3. Modifier encore le code pour qu’il demande en entrée le nombre n de valeurs
dans la Méthode, puis évalue sur 1000 simulations distinctes l’écart-type par
rapport à π.

4. En remarquant que la courbe d’équation y = √1 − x2 est celle d’un demi
cercle de centre O et de rayon 1, comparer le nombre d’itérations nécessaires
pour obtenir une approximation de πde précision comparable avec une mé-
thode de rectangles sur point milieu ou méthode de Monte Carlo.

Remarque : La Méthode de Monte Carlo n’est pas particulièrement efficace pour
calculer les intégrales auxquelles on peut se retrouver confronté en prépa. Elles sont
toutefois concrètement utilisées pour des problèmes de dimensions plus grande, avec
par exemple de nombreux paramètres comme pour des décisions stratégiques, ou en
finance

9.3 Propositions de solutions

Rectangles

Ici, on calcule l’intégrale entre 0 et x, que l’on ajoute ou soustrait à 1
2 en fonction

du signe de x. En effet, on a :



x

−∞

1
√2π

e−
t2
2 dt = 

0

−∞

1
√2π

e−
t2
2 dt+

x

0

1
√2π

e−
t2
2 dt = 1

2+

x

0

1
√2π

e−
t2
2 dt

function y=f(x)
y=exp(-x.^2/2)/sqrt(2*%pi)

endfunction

function y=frepartition(x)
n=500
u=linspace(0,abs(x),n+1)
u=0.5*(u(1:n)+u(2:n+1))
valeurs=f(u)
int=sum(valeurs)*abs(x)/n
if x>=0 then

y=0.5+int
else

y=0.5-int
end
endfunction

La commande integrate ne donne rien de pertinent avec une borne inférieure trop
petite comme −1099. Vous pouvez vous convaincre que −100 suffit en tapant
X=integrate('f(x)','x',-100,1);X
Vous pourrez ensuite comparer par exemple le résultat de
X=integrate('f(x)','x',-100,a);X et de
frepartition(a),
pour a = 1, a = 4, a = −10, et toutes les valeurs que vous voulez.

Tant que l’on reste dans le domaine « utile » (à quelques σde 0), les résultats sont
très précis avec une subdivision de taille 500. Au delà de 5σ, on gagnerait à optimiser
la subdivision en « resserrant » les valeurs autour de zéro. Ce type de découpage à
pas non constant sort du cadre de ces TP.

Monte-Carlo

// approximation de pi
// par Montecarlo
// version vectorielle
n=20000
x=2*rand(1,n)-1
y=2*rand(1,n)-1
r=x.^2+y.^2 // distances à O
proportion=sum(r<1)/n

aire=proportion*4
disp(aire)

Si je trouve le code précédent plus élégant, on peut envisager une version itérative
dans l’optique de modifier le code pour la question 2.

// approximation de pi
// par Montecarlo
// version iterative
n=20000
compteur=0
for i=1:n

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
end
proportion=compteur/n
aire=proportion*4
disp(aire)

Pour la question 2…

// approximation de pi
// par Montecarlo
// nombre d'iterations avant approximation suffisante
precision=0.0001
compteur=0
nbiter=1
while abs(4*compteur/nbiter-%pi)>precision

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
nbiter=nbiter+1

end
disp(nbiter,'le nombre d''itérations nécessaires est : ')

Quelques tests vous montrerons que le résultat est très fluctuant. Notamment,
si on sait que la méthode converge, un bon résultat peut être momentanément
« empiré » par des tests supplémentaires.
Il est certainement plus sensé de procéder à de l’échantillonnage pour évaluer la
méthode, en s’arrêtant au bout d’un nombre arbitraire, mais relativement impor-
tant, de constater la précision atteinte, et de répéter le processus suffisamment
pour pouvoir effectuer des statistiques avec les données.

// évaluation de la précision de la Méthode de Monte Carlo
n=input('nombre d''itérations : ')
echantillon=[]
for i=1:100

compteur=0
for i=1:n

compteur=compteur+(((rand()*2-1)^2+(rand()*2-1)^2)<1)
end
proportion=compteur/n
echantillon=[echantillon,proportion*4]

end
ecart=echantillon-%pi

ecarttype=sqrt(sum(ecart.^2))
disp(ecarttype)

Avec un échantillon de taille 100, j’ai obtenu un écart type de l’ordre de 0,12 pour
10000 itérations, et de l’ordre de 0,03 pour 100000 itérations (ne le lancez que si
vous aimez prendre un café devant votre ordinateur, c’est long…)

Moralité, on peut obtenir un bon résultat très rapidement, mais le nombre d’ité-
rations nécessaires pour obtenir un intervalle de fluctuation suffisamment petit
pour que le résultat soit exploitable est trop important. Aussi, on préférera une
méthode numérique de type « point milieu » ou« Simpson »(très couramment
utilisée).

Pour ce dernier TP, beaucoup de révisions, rappel des commandes que vous devez
connaître, de petits morceaux de code à savoir replacer, et des exercices.

10.1 Principales Commandes
Voici la liste des commandes au programme de première année. Il faut savoir les
utiliser pour écrire ou compléter un programme.

Logique, comparaisons

opérations arithmétiques + - * / ^+ - .* ./ .^

comparaisons, tests
(→if, while)

== > <
>= <= <>

and, or & |

Constantes à connaître

π %pi
e %e

True (booléen) %T
False (booléen) %F

plus petite valeur
connue par Scilab

%eps

(peut utilement remplacer 0)

Tableau 10.1 : Nombres, constantes

Fonctions de référence

xn x^n

√x sqrt(x)
|x| abs(x)

cos(x) cos(x)
sin(x) sin(x)
tan(x) tan(x)
arccos(x) acos(x)
arcsin(x) asin(x)
arctan(x) atan(x)

ex exp(x)
ln(x) log(x)
log10(x) log10(x)

partie entière ⌊x⌋ floor(x)

Hasard

nombre aléatoire dans ]0; 1[ rand()
lois de probabilité grand

Tableau 10.2 : Fonctions

Vecteurs

vecteur ligne u := (1; 2; 3) u=[1,2,3]

vecteur colonne u := ⎛
⎝

1
2
3
⎞
⎠

u=[1;2;3]

dimension d’un vecteur u length(u)

Matrices

M := ⎛
⎝

1 2 3
4 5 6
7 8 9

⎞
⎠

M=[1,2,3;4,5,6;7,8,9]

tM M'
M−1 inv(M) ou M^(-1)
Mn M^n

rang d’une matriceM rank(M)
dimensions d’une matrice size(M)

Tableau 10.3 : Commandes matricielles

liste des variables utilisées who
efface toutes les variables clear

efface la variable x clear(x)

Tableau 10.4 : Quelques commandes utiles

𝐮 = (𝐮u�; ...𝐮u�) étant un vecteur…

n
∑
k=1

uk sum(u)

∑
1≤i≤n

i=1+3p(p∈ℤ)

i2 sum((1:3:n)^2)

n
∏
k=1

sin(k) prod(sin(1:1:n))

n
∏
k=1

uk prod(u)

Tableau 10.5 : Sommes et produits

Pour effacer la fenêtre
graphique en cours

clf()

Pour tracer le graphe d’une
fonction f sur l’intervalle
couvert par le vecteur x

clf();plot(x,f(x))

clf();plot2d(x,f(x))

Mise en forme de données
par exemple : d=rand(1,10000,'normal');

(simule un échantillon gaussien de taille 𝟏𝟎𝟎𝟎𝟎)

diagramme en
barre des données x

clf();x=-2:5;bar(x)

diagramme en barre des
données y en fonction de x

clf();x=-2:5;y=x.^2
bar(x,y,'red')

clf();histplot(20,d)

Tableau 10.6 : Commandes graphiques

10.2 Quelques codes à connaître

Pas de rappels sur la syntaxe ou autre, aucune prétention à l’exhaustivité ici, juste
Mais attention, ce n’est
pas suffisant pour être à
l’abri des surprises.

une succession d’exemples, des commentaires laconiques, pour revoir l’essentiel
en peu de temps.

Attention : Il n’y a ici aucune volonté d’optimisation des algorithmes proposés ici.
Le parti pris est celui de la simplicité, lisibilité, transposabilité du code, souvent
au détriment d’un certain « réalisme » informatique (minimisation du nombre
d’opérations, gestion de la mémoire…). Mais il ne s’agit pas d’un concours d’entrée
en école d’informatique…

Généralités

⋆ Attention à l’indentation, qui fait partie des attendus.

⋆ Écrivez avec une calligraphie propice, sans ambiguïté entre m, n….

⋆ Utilisez les noms de variable de l’énoncé et faites en sorte que vos initiatives
portent sur des noms explicites.

donc attentions aux
habitudes qui sont ren-
forcées en situation de
stress

⋆ Certains caractères ne s’écrivent pas en code, comme les lettres grecques α,
γ, ε… auxquelles vous préférerez alpha, gamma, eps… mais aussi à certains
symboles comme × : qui deviennent * / ou 10−3 qui deviendra 10^(-3) ou
tout simplement 0.001 (et pas de virgule…)

⋆ Attention aux , ; :

⋆ Pensez au double == dans les tests if ou while

⋆ vous pouvez commenter avec //

Quelques sommes

⋆ On calcule n! avec prod(1:n) ou prod([1:n])

⋆ On calcule∑
n
k=1 k avec sum(1:n) ou sum([1:n])

⋆ On calcule∑
n
k=1 k2 avec sum([1:n].^2)

⋆ On calcule∑
n
k=1

1
k avec sum([1:n].^(-1))

⋆ On calcule∑
n
k=0 qk avec sum(q^[0:n])

⋆ On calcule∑
n
k=0 k.qk avec sum([0:n].*q^[0:n])

Retenez le ’point’ qui assure que l’opération entre vecteurs soit réalisée compo-
sante par composante. En particulier .^(-1) pour les inverser les composantes
du vecteur.

Quelques compléments sur les fonctions Scilab

Si l’énoncé demande « Écrire une fonction… », le code doit commencer par
function... et finir par endfunction. Si on vous demande une fonction ré-
cursive, cette fonction devra s’appeler elle-même.
On peut presque tout écrire avec des fonctions si apprécie la tournure, et les
appeler après pour en utiliser le résultat dans un programme.

Qu’est-ce qui différencie une fonction d’un programme

Dans les deux cas, il y a une liste plus ou moins longue d’instructions qui s’en-
chaînent pour réaliser une tâche. Il y a toutefois quelques différences majeures
entre les deux :
⋆ Un programme s’exécute tandis qu’une fonction s’appelle.

Quand vous exécutez un programme, vous exécutez les instructions qu’il
contient dans l’ordre où elles sont écrites. Quand on place une définition
de fonction dans un code Scilab et qu’on exécute ce code, il ne se passe rien
(de visible). On charge alors en mémoire une nouvelle fonction, qui s’ajoute à
la liste des exp et autre floor fonctions de base de Scilab . Une fois chargée
en mémoire, on peut appeler la fonction quand on veut.

⋆ Un programme peut appeler une fonction, une fonction peut appeler une autre
fonction, et même s’appeler elle-même dans le cas d’une fonction récursive
mais une fonction n’exécutera pas de programme, a priori. Écrire une fonction
peut rendre plus claire l’écriture d’un programme qui y fait référence.

⋆ Enfin, l’autre grande nuance se joue au niveau des entrées/sorties, et plus
généralement du statut des variables. Un programme travaille avec des va-
riables globales. Toutes les variables en mémoire au moment de l’exécution du
programme sont à disposition. Toutes les variables crées, ou modifiées par un
programme sont accessibles après l’exécution du programme. C’est pourquoi
on les appelle globales. La gestion des variables par les fonctions sera détaillée
dans le paragraphe suivant.

Entrées et sorties d’une fonction

Une fonction peut demander plusieurs valeurs en entrée et renvoyer plusieurs
variables en sortie. Par exemple :

function [a,b]=exemple(x,y,z)
a=x.^2+y.^2
c=x+y+z
b=(z<sqrt(a))

endfunction

Ici, c est une variable interne. Exécuter le code pour mettre la fonction en mémoire
n’affecte aucune variable du système autre que exemple qui est crée en tant que
nouvelle fonction disponible. Pour appeler la fonction, on tape un des deux codes
suivant :

exemple(1,2,3)
[e,f]=exemple(1,2,3)

La première commande attribue les valeurs 1, 2 et 3 à x, y et z, le temps d’effectuer
les calculs nécessaires pour définir la sortie. Après exécution, x, y et z retrouvent
les valeurs qu’ils avaient avant s’ils étaient définis, ou redeviennent « non défi-
nis » dans le cas contraire. Ce sont des variables internes. Elles ne servent que
de lien. Dans l’exemple de fonction, c ne sert également qu’en interne (et pour
cette fonction, ne sert à rien).

Le premier appel exemple(1,2,3) renvoie uniquement la valeur attribuée à a
lors de l’appel de la fonction (mais a est interne également et n’a pas de valeur.

Le second appel [e,f]=exemple(1,2,3) calcule en interne les sorties a et b et
les attribue aux variables e et f en global lors de l’appel de la fonction

À retenir : pas de input ou de disp dans une fonction.
Ces commandes ne sont présentes que dans les programmes.

Fonction récursive pour une suite

On définit la suite (un) par :

{

u0 = 0
un+1 = un

2 + 1
n+1

Écrire une fonction récursive suite qui prend en entrée n et renvoie un.

function u=suite(n)
if n==0 then

u=0
else

u=suite(n-1) //séparé pour un code plus lisible
u=u/2-1/n // attention 1/n

end
endfunction

Fonction récursive pour une récurrence double

Écrire une fonction suite qui prend un entier n en entrée et renvoie an où (an)
est la suite définie par a0 = 1, a1 = 2 et pour tout n ∈ ℕ, an+2 = −3an+1+an.

function u=suite(n)
if n==0 then

u=1
elseif n==1 then

u=2
else

u=-3*suite(n-1)+suite(n-2)
end

endfunction

Fonction récursive pour calculer les coefficients du binôme

En utilisant l’identité (nk) =
n
k (

n
k), écrire une fonction binom qui à deux entiers

n, k vérifiant 0 ≤ k ≤ n, associe (nk).

function b=binom(n,k)
if k==0 then

b=1
else

b=(n/k)*binom(n-1,k-1)
end

endfunction

10.3 Convergences

Déterminer le plus petit entier n…

On reprend la suite (un) par u0 = 0 et un+1 = un
2 + 1

n+1 .
On suppose programmée une fonction suite qui calcule un quand on lui donne
n en entrée et on admet que limn→+∞ n.un = 2.

Écrire un programme qui détermine le plus petit entier n tel que |nun−2| < 10−3

n=0
while abs(n*suite(n)-2)>=0.001

n=n+1
end
disp(n)

L’essentiel des programmes de type while avec une suite enregistrée dans une
fonction suite peuvent se coder ainsi.

Quelques remarques :
⋆ Si le problème parle de deux suites (un) et (vn) adjacentes, et que la valeur de la

limite n’est pas connue, on peut utiliser comme test d’arrêt v-u > precision
on peut alors prendre comme valeur approchée à precision près de la limite
la moyenne (u+v)/2 des termes de la suite auxquels on est arrivés.

⋆ si votre suite est monotone, on peut omettre le abs

Une somme

Écrire une fonction qui demande un nombre n et un nombre x puis affiche Sn =
n
∑
k=0

xn
n!

function s=somme(x,n)
if n==0 then

s=1
else

s=x^n/prod(1:n)+somme(x,n-1) // prod(1:n) calcule n!
end

endfunction

n=input('Entrez un nombre n : ')
x=input('Entrez un nombre x : ')
disp(somme(x,n))

10.4 Simulation de variables aléatoires classiques
Les commandes suivantes renvoient une matrice m × n dont les coefficients
suivent la loi …
⋆ Loi uniforme entière : grand(m,n,"uin",n,p) coefficients dans [[n; p]].

⋆ Loi binomiale ℬ(N; p) : grand(m,n,"bin",N,p)

⋆ Loi de Bernoulli de paramètre p : ℬ(1; p) : grand(m,n,"bin",1,p)

⋆ Loi géométrique 𝒢(p) : grand(m,n,"geom",p)

⋆ Loi de Poisson 𝒫(λ)d’espérance λ : grand(m,n,"poi",lambda)

⋆ Loi uniforme continue sur [a; b[ : grand(m,n,"unf",a,b)

⋆ Loi exponentielle ℰ(λ) : grand(m,n,"exp",1/lambda) dernier paramètre =
espérance

⋆ Loi normale 𝒩(μ, σ2) : grand(m,n,"nor",mu,sigma)
Pour simuler des lois non-standards, il peut être utile de savoir simuler « à la
main » certaines de ces variables aléatoires, notamment les discrètes. On peut
utiliser rand() qui renvoie un nombre aléatoire de [0; 1[ que l’on peut comparer
à la probabilité p pour simuler un succès/échec.
Toutes ces commandes peuvent au besoin être intégrées dans une fonction Scilab.
⋆ simuler une épreuve de Bernoulli de paramètre p. X=(rand()<p) renvoie 1

avec une probabilité p et 0 sinon.

⋆ simuler n épreuves de Bernoulli de paramètre p. X=(rand(n,1)<p) renvoie
un vecteur aléatoire

⋆ simuler ℬ(n; p) : X=sum(rand(n,1)<p)

⋆ simuler une loi géométrique :

X=1
while (rand()>=p)

X=X+1
end

⋆ simuler une loi uniforme sur [a; b] : X=a+(b-a)*rand()
Après chacun de ces codes, on peut afficher la valeur du résultat obtenu avec
disp(X)mais il peut être plus avantageux de créer une fonction Scilab qui simule
la variable aléatoire.

10.5 Résolution d’équations, dichotomie
La dichotomie n’est pas la seule méthode, mais les autres méthodes, par des suites
du type un+1 = f(un) découlent d’éléments vus précédemment. La résolution
approchée par dichotomie est un algorithme à part, que nous allons illustrer
avec l’approximation de la solution de l’équation (x2 + 1) e−x − x = 0 dont
on peu prouver qu’elle admet une unique solution dans comprise entre 0,5 et 1.
L’algorithme suivant en donne une valeur approchée à 10−6 près :

function y=f(x)
y=(x^2+1)*exp(-x)-x

endfunction

a=0.5
b=1
while b-a>=10^(-6)

c=(a+b)/2
if f(a)*f(c)>0 then

a=c
else

b=c
end

end

disp(a)

10.6 Exercices

Exercice 72 : EML 2010 (ECS)
On note f : ℝ → ℝ l’application définie par : ∀x ∈ ℝ, f(x) = x − ln(1 + x2)
On considère la suite la suite (un) définie par u0 = 1 et un+1 = f(un).
1. Écrire un programme qui calcule et affiche le plus petit entier n tel que
un ≤ 10−3

2. a. Établir ∀x ∈ [0; 1], f(x) ≤ x − x2

2
b. Déduire ∀n ∈ ℕ, u2

n ≤ 2(un − un+1)
c. Montrer que la série ∑

n≥0
u2
n converge et que

∀n ∈ ℕ,
+∞
∑

k=n+1
u2
k ≤ 2un+1

d. Écrire un programme qui calcule et affiche une valeur approchée de
+∞
∑
n=0

u2
n

à 10−4 près.

Exercice 73 L’urne U1 contientN boules : (N − 1) boules blanches et 1 boule
noire. On effectue des tirages sans remise dans l’urne U1 , jusqu’à l’obtention de
la boule noire. On note X la variable aléatoire qui prend pour valeur le nombre de
tirages nécessaires pour l’obtention de la boule noire.
1. On simule 10000 fois cette expérience aléatoire. Recopier et compléter le pro-

gramme Scilab suivant pour qu’il affiche l’histo- gramme donnant la fréquence
d’apparition du rang d’obtention de la boule noire :

N=input('Donner un entier naturel non nul');
S=zeros(1,N)
for k=1:10000

i=1;
M=N;
while _ _ _ _ _ _ _ _ _ _ _ _ _ _

i=i+1
M= _ _ _ _ _ _ _ _ _ _ _ _ _ _

end
S(i)=S(i)+1;

end
disp(S/10000)
bar(S/10000)

2. On exécute le programme complété ci-dessus. On entre 5 au clavier et on
obtient l’histogramme suivant :

Quelle conjecture pouvez-vous émettre sur la loi de la variable aléatoire X ?

Exercice 74
On lance n fois une pièce de monnaie équilibrée, on note Y la variable aléatoire
qui vaut 0 si pile tombe au plus une fois, et qui dans le cas contraire prend pour
valeur le rang du deuxième pile.

1. Écrire une fonction Scilab nommée varY permettant de simuler la variable
aléatoire Y

2. Écrire une fonction Scilab nommée table prenant en entrée un entier n
correspondant au nombre de tirages, et qui renvoie un vecteur contenant le
nombre d’occurrence de chaque valeur. (la première coordonnée contiendra
ℙ(Y = 0), la seconde ℙ(Y = 2) … la k-ième ℙ(Y = k)

3. En déduire une approximation de 𝔼(Y)

penser à
usecanvas(%T)

4. Afficher un diagramme en barres présentant les fréquences d’apparition de
chaque valeur.

Index

a
aide 5
ans 5

b
booléens 7, 8

c
chaîne de caractères

concaténation 9
définition 9

e
égalité

affectation 8
test 8

f
fonctions de référence 5, 6

g
graphiques 11

l
logique 8

m
matrices 10

p
puissances 10

r
représentation graphique 11

s
Scilab 5
sommes 11

v
variables

affectation 6
vecteurs 10, 11

Index des commandes

a
abs 6
ans 5

c
clf() 11
cos 6

e
exp 6

f
floor 6

h
help 5

l
length 10

log 6
log10 6

p
part 10

r
rand 6

s
sin 6
sqrt 6
sum 10, 11

t
tan 6
plot 11
& 7
, 7, 10

.. 7
// 7
; 7, 10
= 7
== 7, 8
| 7
~ 7

u
usecanvas 11

w
who 6




